
INTRODUCTION 

This document defines the Allied Motion Programmability Standard (AMPS). 

The purpose of this document is: 

1. To define a motion language and an associated run-time that is used in all programmable drives across 

TU’s that: 

a. Encourages commonality of firmware and features; 

b. Minimizes development time and cost; 

c. Minimizes customer support while enhancing familiarity through use of a simple but powerful 

language and programming/commissioning tools; 

d. Is extensible and flexible enough to meet the widest variety of customer applications; 

e. Can be easily extended for multiple axis operation. 

2. To inspire discussion and communication among drive design and application engineers. 

CONCEPT 

In general, user-programmable drives consist of three components: 

1. A means to describe the intended logic and desired motion, 

2. A means to translate the aforementioned logic and motion into a form suitable for execution in the drive, 

3. A firmware run-time system to execute the translated logic and motion. 

For the AMPS implementation, these components are: 

1. A programming project containing one or more “program organizational blocks” (POBs) written in a “C”-

like, text-based programming language (AML), 

2. A translator that converts the project’s POBs to a series of opcodes (“operation codes”) suitable to be 

executed by the firmware-based run-time, 

3. A virtual machine run-time (VM) that interprets and executes the opcodes. 

The diagram on the following page illustrates the implementation. 

TRAJECTORY
GENERATOR

enable drive;
[loop]
move at: 10 for: 1;
wait 5000 ms;
move at: 10 for: -1;
wait 5000 ms;
goto loop;

User Program (Program Organizational Blocks)

MULTI-TASKING
VIRTUAL

MACHINE
(VM)

/* 000000 */ 0x0600,0x0000,                    /* STACK */
/* 000004 */ 0x0602,0x0046,0x0000,             /* DRAD $$$DATA */
/* 00000A */ 0x0404,                           /* DREN */
/* 00000C */                                   /* [LOOP] */
/* 00000C */ 0x0100,0x000A,0x0000,             /* PUSHCI 10 */
/* 000012 */ 0x0100,0x0001,0x0000,             /* PUSHCI 1 */
/* 000018 */ 0x0012,0x0011,                    /* CALLX */
/* 00001C */ 0x0100,0x1388,0x0000,             /* PUSHCI 5000 */
/* 000022 */ 0x0407,                           /* DELAY */
/* 000024 */ 0x0100,0x000A,0x0000,             /* PUSHCI 10 */
/* 00002A */ 0x0100,0x0001,0x0000,             /* PUSHCI 1 */
/* 000030 */ 0x0210,                           /* NEG */
/* 000032 */ 0x0012,0x0011,                    /* CALLX */
/* 000036 */ 0x0100,0x1388,0x0000,             /* PUSHCI 5000 */
/* 00003C */ 0x0407,                           /* DELAY */
/* 00003E */ 0x0300,0x000C,0x0000,             /* BR LOOP */
/* 000044 */ 0x0401,                           /* END */
/* 000046 */                                   /* [$$$DATA] */
/* 000046 */ 0x0603,0x004C,0x0000              /* DATA $$$END */
/* 00004C */                                   /* [$$$END] */

User writes one or more Program 
Organizational Blocks (POBs) in a AML 
using IN Control

A translator (part of IN Control) converts to 
each POB to an intermediate “opcode” 

language

The Virtual Machine (VM) schedules, triggers 
and runs one or more POBs.  

The VM sends motion commands and data as 
needed to the trajectory generator

PROGRAM ORGANIZATIONAL BLOCKS 

Program Organizational Blocks (POBs) are the components that make up a user program.  Typically, more than one 

POB is used to construct the user program, although only one is required. 

POBs operate and execute independently of each other – in other words, they also can be thought of as tasks in a 

multi-tasking environment. 

POBs have priority – namely, foreground (FG) and background (BG).  FG POBs are scheduled to run with the 

position loop and as such should be limited to critical operations (such as starting an indexed move when an input 

changes state). There is a limit to the number of active POBs can be run in the foreground before an error occurs. 

The run-time virtual machine also detects when the active FG POBs have run for too long (which would create a 

problem scheduling the next position loop update).  This is a fatal error as well.  Computationally expensive 

operations, such as trigonometric math, should be computed in a BG POB and passed somehow to the FG POB. 

POBs can start, stop and idle (suspend) other POBs, both FG and BG. 

POBs can be set to start running when some event occurs – for example, when the state of an input changes.  This 

event-driven programming is very useful in handling both expected and unexpected events in automated systems.  

For example, instead of using the drives pre-programmed over-travel limit inputs, the user can program event 

triggers to run when the limit switch is tripped to gracefully stop the motor, annunciate such to the operator and 

to restart operation in a logical manner (such as entering a jog mode). 

As mentioned earlier, at least one POB is required in a user program.  This required POB must be named main.  

This POB typically initializes the system, and starts the necessary POBs required for operation of the drive.  Main is 

started as a BG POB. 

There is only one instance of a running POB – if a POB named homeAxis is running and a POB issues a command to 

start it (again), the command will be effectively ignored. 

In summary: 

1. POBs are the components that make up a user program in the drive; 

2. POBs are either run with the position loop (FG POBs) or run when no other loops are scheduled (in the 

free time as BG POBs); 

3. FG POBs are resource limited – a limited number can be active at one time, and excessive CPU use in FG 

POBs will cause a fatal error (“FG POB Time Exceeded”); 

4. FG can be started, stopped and idled by another POB and can be triggered to start when an event occurs; 

5. At least one POB is required in the user program – this POB named main initializes the system and starts 

the execution of other POBs as required. 

MULTI-AXIS PROGRAM ORGANIZATION BLOCKS 

In a multi-axis environment, a user program consists of: 

1. A supervisory set of POBs destined for the supervisory controller (“axis 0”); 

a. This set of POBs must contain the main POB that initializes the entire multi-axis system. 

2. Zero or more POBs for each axis in the system. 

The supervisory controller governs the entire operation of the multi-axis system and as such can: 

1. Start, stop or idle pre-stored POBs on a given axis; 

2. Send on-the-fly AML code to one or more axes for execution. 

The supervisory controller has access to the global variables of each axis as well as its own store of global variables.  

An axis can access global variables on the supervisory controller, but not those of other axes. 

VARIABLES 

The VM run-time supports global, local and system variables. 

All global and local variables are typed at run-time by the VM.  In other words, storing an integer into a variable 

temporarily makes that variable an integer.  If one later stores a floating-point value into the same variables, the 

variable is then typed as a floating-point. 

Global variables are accessible from all POBs as well as through communication means.  Thus, global variables can 

be read and written to by using IN Control as well as through fieldbuses such as CANopen and EtherCAT. 

Local variables are accessible only within the POB they appear.  They are inaccessible from the outside world, 

although they can be viewed by using IN Control for debug purposes.  Local variables begin with the “_” character 

to identify them clearly as locals. 

A POB can access parameters and feedback variables (collectively known as system variables) associated with the 

drive such as the velocity-loop tuning parameters or the feedback velocity (VEL).  No special prefix is used with 

these system variables. 

⚠ A run-time fault occurs if an attempt is made to assign a value to a read-only system variable. 

All variables are case-insensitive; therefore “xyz” and “XYZ” refer to the same variable. 

 

ARRAY VARIABLES  

The VM run-time supports global array variables of one or two dimensions. 

One-dimensional arrays are referred to as vectors and are named starting with VEC followed by a digit.  Two-

dimensional arrays are referred to as tables and are named starting with TBL followed by a digit. 

Arrays may be read from and written to by means of the vector operator ([index]) or the table operator 

([index,index]). 

Vectors and tables may be dynamically allocated or re-allocated using AML. 

 

SCHEDULING AND ATOMICITY 

Regardless of whether a POB is prioritized as FG or BG, they are executed in a round-robin manner. 

FG POBs are executed in a round-robin fashion within the position loop, whereas BG POBs are executed in a round-

robin manner when no other loops are running (“in the idle time”). 

When a POB is active (running), a single AML statement is executed before relinquishing to the next active POB to 

be run.  In other words, if an FG POB contains three assignment statements in a row, only one will execute in this 

position-loop update – the subsequent two statements will execute the next time (the second assignment 

statement) and the time following (the third assignment statement). 

There are exceptions to this rule, however.  Certain AML statements intentionally bypass the round-robin rule.  

Those statements that do are highlighted later in this document (in the AML language specification) – they typically 

consist of commands related to position capture (registration) or gearing where performance is required. 

The user can also specify that the VM run-time temporarily suspend the round-robin rule through use of an atomic 

block in AML code.  If, instead, the aforementioned three assignment statements where located in an atomic 

block, they would be executed in one position-loop update. 

⚠ Care must be taken when using atomic blocks in FG POB code as more CPU resources are utilized and the risk of 

an error increases (“FG POB Time Exceeded”). 

⚠ Certain AML statements will force a round-robin switch regardless of whether they occur within an atomic 

block.  This is to prevent a single POB from monopolizing time from other POBs.  Those statements that force a 

round-robin switch will be defined later in this document. 

AML 

AML is the “C”-like imperative language that POBs are programmed with.  POBs contain a series of AML 

statements. 

Included in the language are the common “C” constructs, such as the structured programming “if statement”, as 

well as non-“C” motion and control -oriented statements. 

AML SYNTAX 

A semicolon (“;”), the same as in “C”, terminates a statement in AML.  Block type statements (such as atomic) do 

not use a trailing semicolon to terminate, just as in “C”. 

Comments begin with two forward slashes (“//”) and continue to the end of the line as in “C++”.  AML also 

supports /* to */ commenting as well the same as in “C” and “C++”. 

Statements are entered in a free-form manner allowing for arbitrary use of whitespace to format code. 

AML EXPRESSIONS 

Expressions, like in “C”, are fundamental in AML.  Expressions consist of numeric constants (such as 0, 1.3 and -

10.5E-6), variables (global/system or local), function calls (such as sin and abs) and operators (such as +, * and <=). 

 

AML OPERATORS 

AML operators have precedence similar to “C”. 

Symbol Operator Operation 

( expression ) Parenthesis  

[ index ] Vector access Vector subscripting (1-dimensional array) 

[ index , index ] Table access Table subscripting (2-dimensional array) 

var@axis Axis variable access Used to interrogate the value of a variable on another axis 

object'field Object access Used to interrogate an object 

+ Unary plus + X returns X 

- Unary minus - X returns the 2s complements of X (negate) 

! Logical “not” ! X returns “1” if X is 0, “0” otherwise 

~ Bit-wise “not” ~ X returns the 1s complement of X (bit inversion) 

# Bit access X # n returns “1” if the nth bit (0=LSB) in X is set, “0” otherwise 

* Multiply  

/ Divide  

% Modulo X % Y returns the remainder of X divided by Y as an integer 

+ Add  

- Subtract  

<= Less than or equal X <= Y returns “1” if X <= Y, “0” otherwise 

< Less than X < Y returns “1” if X < Y, “0” otherwise 

> Greater than X > Y returns “1” if X > Y, “0” otherwise 

>= Greater than or equal X >= Y returns “1” if X >= Y, “0” otherwise 

!= Not equal X != Y returns “1” if X and Y are different in value, “0” otherwise 

== Equal X == Y returns “1” if X and Y are equal in value, “0” otherwise 

& Bit-wise “and” X & Y returns the bit-wise “and” of X and Y 

| Bit-wise “or” X | Y returns the bit-wise “or” of X and Y 

&& Logical “and” X && Y returns “1” if both X or Y are non-zero, “0” otherwise 

|| Logical “or” X || Y returns “1” if either X or Y are non-zero, “0” otherwise 

AML FUNCTIONS 

 

Function Operation 

abs(x) Absolute value of x 

acos(x) The inverse cosine of x (returned in radians) 

asin(x) The inverse sine of x (returned in radians) 

atan(x) The inverse tangent of x (returned in radians) 

atan2(y,x) The angle (in radians) between the positive x-axis of a plane and the point given 
by the coordinate (x,y) 

ceil(x) The smallest integer not less than x 

cos(x) The cosine of x (where x is in radians) 

exp(x) ex 

fallof(x) Returns 1 if input #x transitions from on to off, 0 otherwise 

floor(x) The largest integer no greater than x 

log(x) The natural logarithm of x 

log10(x) The logarithm of x, base 10 

pow(x,y) xy 

riseof(x) Returns 1 if input #x transitions from off to on, 0 otherwise 

sgn(x) Sign of x (-1 for x < 0, 0 for x == 0, 1 for x > 0) 

sin(x) The sine of x (where x is radians) 

sqrt(x) The square root of x 

tan(x) The tangent of x (where x is in radians) 

BASIC STATEMENTS 

 

Assignment 
syntax variable = expression 

arguments variable The variable to store the value of expression into 
expression A numeric expression to evaluate 

example // increment z 
z = z + 1; 
// compute the length of the index 
index_length = 3.5 + 2*z; 

The assignment statement changes the value of a variable. 

 

If / Set 
syntax if condition set variable = expression 

arguments condition A expression to test; the assignment occurs if the condition evaluates to a 
non-zero value 

variable The variable to store the value of expression into 
expression A numeric expression to evaluate 

example // clear x when x >= 10 
if x >= 10 set x = 0; 

The if/set statement performs an assignment when a specified condition is true.  Since this is a common operation, 

this statement is essentially a short form of the more complicated if/then statement and makes AML code easier 

to read and understand. 

 

If / Then 
syntax if condition then { statement; statement; … statement; } 

arguments condition A expression to test 
{ … } A bracketed list of statements to execute if the expression evaluates to a 

non-zero value 
example // if the first input is ON, clear x and y 

if INS#1 then { 
 x = 0; 
 y = 0; 
} 

The if/then statement performs a series of statements when a specified condition is true.  If the condition 

evaluates to false (a zero value), then these statements are ignored and not executed. 

If/then statements may be nested – in other words, within the bracketed list of statements to execute, another 

if/then or if/then/else statement is allowed. 

If / Then/Else 
syntax if condition then { statement; … statement; }

1
 else { statement; … statement; }

2
 

arguments condition A expression to test 
{ … }

1
 A bracketed list of statements to execute if the expression evaluates to a 

non-zero value 
{ … }

2
 A bracketed list of statements to execute if the expression evaluates to a 

zero value 
example // if the first input is ON, clear x and y 

if INS#1 then { 
 x = 0; 
 y = 0; 
} 
// otherwise, increment z 
else { 
 z = z + 1; 
} 

The if/then/else statement performs a series of statements when a specified condition is true and a different series 

of statements when the specified condition is false (a zero value). 

If/then/else statements may be nested – in other words, within the bracketed list of statements to execute, 

another if/then or if/then/else statement is allowed. 

 

Reset 
syntax reset 

example // reset faults 
reset; 

The reset statement attempts to reset any controller or run-time faults. 

It is not an error to reset a non-faulted controller. 

 

Allocation (Vector) 
syntax alloc variable [size] 

arguments variable A vector variable (VECn) 
size An expression that evaluates to a non-negative size 

example // allocate VEC1 
alloc VEC1[100]; 
// use VEC1 
VEC1[0] = 123.4; 
VEC1[1] = VEC1[0] + 456.7; 
// ... 
// de-allocate it 
alloc VEC1[0]; 

The alloc (vector) statement allocates (or reallocates or de-allocates) storage for a vector variable. 

Any data contained in the vector variable prior to executing the alloc statement is lost. 

⚠ An alloc instruction may only be executed in a background POB. 

Allocation (Table) 
syntax alloc variable [rows,cols] 

arguments variable A table variable (TBLn) 
rows An expression that evaluates to a non-negative number of rows 
cols An expression that evaluates to a non-negative number of columns 

example // allocate TBL2 
alloc TBL2[10,2]; 
// use TBL2 
TBL2[0,0] = 123.4; 
TBL2[0,1] = TBL2[0,0] + 456.7; 
// ... 
// de-allocate it 
alloc TBL2[0,0]; 

The alloc (table) statement allocates (or de-allocates) storage for a table variable. 

Any data contained in the table variable prior to executing the alloc statement is lost. 

⚠ An alloc instruction may only be executed in a background POB. 

 

Link (Variable) 
syntax link dest = var1 * con1 + var2 * con2 + offset 

link dest = var1 * con1 + var2 * con2 - offset 
link dest = var1 * con1 
link dest (lower,upper) = var1 * con1 + var2 * con2 + offset 
link dest (lower,upper) = var1 * con1 + var2 * con2 - offset 
link dest (lower,upper) = var1 * con1 

arguments lower An optional lower limit (constant) 

upper An optional upper limit (constant) 
dest A (writeable) destination system variable to link 
var1 A variable 
con1 A constant 
var2 An optional variable 
con2 An optional constant 
offset An optional constant 

example // Have ADC1 act as a “feed-rate override” 
link VRATE (0,2) = ADC1 * 0.2; 

The link statement is essentially an assignment statement that is executed repeatedly. 

In the example above, the link statement relates the VRATE system variable with the value of the first analog input.  

When the first analog input changes, VRATE is recalculated and updated. 

 

Unlink (Variable) 
syntax unlink dest 

arguments dest A destination system variable to unlink (or “all”) 

example // Have ADC1 act as a “feed-rate override” 
link VRATE (0,2) = ADC1 * 0.2; 
// ... 
// No longer relate VRATE to ADC1 
unlink VRATE; 

The unlink statement undoes a link operation.  A dest of “all” unlinks all active links. 

Capture (Variable) 
syntax capture ( capturelist ) mode 

capture every: n ( capturelist ) mode 
capture depth: d position: p ( capturelist ) mode 
capture every: n depth: d position: p ( capturelist ) mode 
capture abort 

arguments capturelist A comma-delimited list of up to 4 capture-capable variables 
mode auto    capture immediately 

utrig    capture based upon trigger next statement 
when var >= value  capture when var >= a constant value 
when var <= value  capture when var <= a constant value 
when var outside (min,max) capture when var is outside 2 limits 
when var within (min,max) capture when var is within 2 limits 

n How often to capture; a value of 1 means every 250us, a value of 2 means 
every 500us, etc. 

d How many times to capture the capturelist (1 to 10000) 
p The ‘trigger’ position: 

 a value of 0 to indicate all data should be ‘pre-trigger’ 
 a value of ‘d’ to indicate all data should be ‘post-trigger’ 
 a value > 0, but less than ‘d’ to mix ‘pre’ and ‘post’ -trigger 

example // Set up capture on a fault condition for this axis 
// --- capture target/feedback position information leading up to a fault 
capture depth: 10000 position: 0 (TPOS, FPOS) when FAULTED >= 0; 
// ... 
// ... 
// Capture axis 3 ADC1 when triggered by a trigger next 
capture (ADC1@3) utrig; 

The capture statement in the first four forms sets up a POB-based (“offline”) data capture operation.  The last form 

(“capture abort”) disables POB-based capture. 

POB-based data capture stores the requisite information to the on-board file system for later retrieval by IN 

Control or through the ALLNET framework. 

Up to 4 variables can be captured at the same time.  Any subsequent (non-abort) capture replaces the existing 

capture operation. 

If a POB-based capture (through this statement) is active, no IN Control –based capture is allowed. 

PROGRAM FLOW STATEMENTS 

 

Label 
syntax [name] 

arguments name The name of the label (used as a destination for goto) 

example // define a label so we can go to it 
[loop] 
x = x + 1; 
if x < 10 goto loop; 
if y > 100 goto next; 
… 
[next] 
… 

A label in AML is a marker for the destination of a goto. 

Label names are case-insensitive and are local to the POB where they are used. 

Label names must be unique within the POB where they are used.  Labels are not allowed in atomic blocks since 

goto-type instructions are not allowed within such. 

 

Goto 
syntax goto name 

arguments name The name of the destination label 

example // define a label so we can go to it 
[loop] 
… 
goto loop; 

The goto statement transfers program flow to a label within the POB. 

⚠ The goto statement causes a round-robin POB switch regardless of where it appears.  The goto statement is not 

allowed in an atomic block. 

If / Goto 
syntax if condition goto label 

arguments condition A expression to test; the branch to label occurs if the condition evaluates 
to a non-zero value 

label The destination to go to when the condition evaluates to true (a non-zero 
value) 

example // define a label so we can go to it 
[loop] 
x = x + 1; 
if x < 10 goto loop; 
 
// test if the first input is ON 
if INS#1 goto first_input_is_on; 
 

The if/goto statement transfers program flow to a label within the POB when the specified condition evaluates to a 

non-zero value. 

⚠ Execution of the goto part of this statement (in other words, when the condition is true) causes a round-robin 

POB switch.  The if/goto statement is not allowed in an atomic block. 

 

Wait 
syntax wait time ms 

wait forever 
arguments time An expression that will be evaluated and used as the number of milliseconds 

to wait (a value of 0 = forever) 
example // wait for 100 milliseconds 

wait 100 ms; 

The wait statement delays for a specified number of milliseconds or forever.  The use of wait/forever is useful in 

conjunction with asynchronous event handlers. 

⚠ Suspending a POB executing a wait instruction effectively suspends the wait timer as well.  In other words, if the 

POB was executing a wait of 100 ms duration and the POB was suspended after 75 ms, then the wait will continue 

for 25 ms after the POB is un-suspended. 

⚠ A wait statement cannot appear within an atomic block. 

 

Wait/Until 
syntax wait until condition 

arguments condition An expression to test 

example // wait until the first and second inputs are ON 
wait until INS#1 && INS#2; 

The wait/until statements stops program flow until the specified condition evaluates to true. 

⚠ Suspending a POB executing a wait/until instruction effectively suspends evaluation of the condition during the 

suspension. 

⚠ A wait/until statement cannot appear within an atomic block. 

Wait/Until/After 
syntax wait until condition after time ms goto label; 

arguments condition An expression to test 

time An expression to evaluate as a timeout value 
label The label to branch to if the condition fails to evaluate to true after time 

milliseconds 
example // wait until the first and second inputs are ON 

// give up after 100 milliseconds 
wait until INS#1 && INS#2 after 100 ms goto not_on; 

The wait/until/after statements stops program flow until the specified condition evaluates to true or after a 

specified period of time.  If the condition is not met within the time limit, program execution continues at the 

specified label. 

⚠ Suspending a POB executing a wait/until instruction effectively suspends evaluation of the condition during the 

suspension as well as suspending the timeout timer. 

⚠ A wait/until/after statements cannot appear within an atomic block.  

POB CONTROL STATEMENTS 

 

Start/Restart POB 
syntax start name FG 

start name BG 
restart name FG 
restart name BG 

arguments name The name of the POB to start (or restart) 

example // home the axis, wait until POB has ended 
start HOME FG; 
wait until HOME’notRunning; 
// restart the UTIL POB 
restart UTIL FG; 

The start statement starts execution of the named POB as either a foreground (FG) or background (BG) POB.  If the 

POB is currently running or suspended, the start will be effectively ignored. 

The restart statement first stops the named POB (if running) and then starts it as either a foreground (FG) or 

background (BG) POB. 

 

Stop POB 
syntax stop name 

stop all 
arguments name The name of the POB to stop 

example // abort homing the axis 
stop HOME; 
// stop all POBs (excluding this one) 
stop all; 

The stop statement stops execution of the named POB or all POBs (excluding the POB executing the stop/all). 

If the named POB is not running, the stop statement will be effectively ignored – suspended POBs are stopped, 

however. 

Executing a stop all in a multi-axis environment stops all POBs running on all axes. 

 

Suspend POB 
syntax suspend name 

arguments name The name of the POB to suspend 

example // suspend the POB named UTIL 
suspend UTIL; 

The suspend statement temporarily suspends a POB.  If the POB is not currently running or suspended, no error 

occurs and the statement is effectively ignored. 

Unsuspend POB 
syntax unsuspend name 

arguments name The name of the POB to unsuspend 

example // unsuspend the POB named UTIL 
suspend UTIL; 

The unsuspend statement reactivates a suspended POB.  If the POB is not suspended, no error occurs and the 

statement is effectively ignored. 

 

End (POB) 
syntax end 

example // end this POB after suspending UTIL 
suspend UTIL; 
end; 

The end statement stops the execution of the POB executing the statement. 

An implicit end statement is added to the end of any POB prior to translation from AML to opcode form. 

IO CONTROL STATEMENTS 

 

Set Output 
syntax set output state ( list ) … state ( list ) 

arguments state ON/on or OFF/off 
list A parenthesized, comma-delimited list of outputs 

example // activate outputs 2 and 3 and deactivate output 4 
set output on (2, 3) off (4); 
// deactivate output 1 
set output off (1); 

The set output statement activates or deactivates a list of outputs. 

 

Pulse Output 
syntax pulse output n for time ms 

arguments n The output to pulse 
time An expression that evaluates to the number of milliseconds to pulse the 

output 
example // pulse output 3 for 50 milliseconds 

pulse output 3 for 50 ms; 
// pulse output 2 for xpulse milliseconds 
pulse output 2 for xpulse ms; 

The pulse output statement pulses an output on for a specified duration. 

If the output is on when the statement executes, the output state will initially be kept on, but will turn off after the 

specified duration has elapsed. 

If a set output statement changes the state to off before the duration elapses, the generation of the pulse is 

effectively aborted. 

 

Define PLS 
syntax define pls n using refvar defs 

arguments n The output to define or replace the definition of 
defs A comma-delimited list of definitions of the form: 

 on const1 to const2 
example // define a PLS for output 1 

// the output will be active (on) when FPOSC is between 0 to 400 or between 900 and 4000 
// the output will be inactive (off) when FPOSC is outside those bounds 
define pls 1 using FPOSC on 0 to 400, on 900 to 4000; 

The define pls statement defines (or redefines) a software-based, programmable limit switch (PLS). 

If a PLS definition is in place for a given output, the definition is replaced by the new definition. 

When a define pls statement is encountered, the PLS is disabled and will not be used to compute the state of the 

output.  An enable pls statement is required to activate the PLS. 

Enable/Disable PLS 
syntax enable pls n 

disable pls n 

arguments n The output to enable/disable the PLS for 

example // enable the PLS defined for output 1 
enable pls 1; 
// disable the PLS defined for output 2 
disable pls 2; 

The enable pls activates the defined PLS for the specified output. 

 

Wait For Rise/Fall 
syntax wait for rise of n 

wait for fall of n 
arguments n The input to wait for 

example // wait until input 1 transitions from off to on 
wait for rise of 1; 
// wait until input 2 transitions from on to off 
wait for fall of 2; 

The wait for rise/fall statement waits for a transition of a digital input. 

⚠ A wait rise/fall statement cannot appear within an atomic block. 

BLOCK STATEMENTS 

 

Atomic 
syntax atomic { statement; … statement; } 

example // save away the previous value of x, and clear x 
atomic { 
 x_old = x; 
 x = 0; 
} 

The use of an atomic block temporarily suspends the round-robin execution of POBs guaranteeing that the 

statements within the atomic block are executed without any other statements (contained in either foreground or 

background POBs) begin executed. 

Use of the atomic statement in a FG POB causes the following to occur: 

1. The FG POB executing the atomic statement is temporarily stopped.  Since FG POBs are effectively 

interrupt driven (as they are run in the position-loop task), there may be BG POBs that are in process of 

being executed and they must be stopped first. 

2. A flag is set in the VM that manages the BG POBs informing that the BG POBs should be temporarily 

stopped after they have finished executing their respective current statements. 

3. The VM managing the BG POBs will set another flag informing the FG POB manager that all BG POBs have 

been temporarily stopped. 

4. The FG POB that was temporarily stopped due to execution of the atomic statement executes the 

statements contained within the atomic block to completion. 

5. The FG POB manager clears all flags allowing the BG (and other FG) POBs to run again. 

Use of the atomic statement in a BG POB causes the following to occur: 

1. A flag is set informing the FG POB manager not to execute any FG POBs. 

2. A flag is set informing the BG POB manager to only run this POB. 

3. When all POB statements within the atomic block have completed execution, all flags are cleared 

resuming normal operation. 

⚠ The atomic statement cannot be nested.  

Within (timeout) 
syntax within time ms { statement; … statement; } else goto label 

arguments time An expression that evaluates to the timeout (in milliseconds) 

label The label to go to if a timeout occurs 
example // wait for the registration mark on input 1 to be “seen” 

// if not “seen” within 200 milliseconds, go to nomark 
within 200 ms { 
 wait for rise of 1; 
 wait for fall of 1; 
} else goto nomark; 

The within statement executes a series of statements with a timeout check. 

If the execution of the statements does not finish within the specified time, program flow is redirected to the 

specified label. 

⚠ Within statements cannot be nested and may not contain goto, call, wait/until/after or label statements or an 

atomic block. 

⚠ Within statements may not appear within an atomic block. 

 

Within (timeout) 
syntax within time ms { statement; … statement; } else { statement; … statement; } 

arguments time An expression that evaluates to the timeout (in milliseconds) 
label The label to go to if a timeout occurs 

example // wait for the registration mark on input 1 to be “seen” 
// if not “seen” within 200 milliseconds, set failed to 1 
within 200 ms { 
 wait for rise of 1; 
 wait for fall of 1; 
} else { failed = 1; } 

The within statement executes a series of statements with a timeout check. 

If the execution of the statements does not finish within the specified time, program flow is redirected to the else 

block. 

⚠ Within statements cannot be nested and may not contain goto, call, wait/until/after or label statements or an 

atomic block. 

⚠ Within statements may not appear within an atomic block. 

Repeat 
syntax repeat count { statement; … statement; } 

arguments count An expression that evaluates to the number of times to execute the 
statements in the block 

example // wait for 5 transitions of input 1 
repeat 5 { 
 wait for rise of 1; 
 wait for fall of 1; 
} 

The repeat statement executes a series of statements a certain number of times. 

If the count evaluates to a negative number or zero, no statements are executed. 

Repeat statements may not appear in an atomic block, but may be nested. 

 

Repeat 
syntax repeat until condition { statement; … statement; } 

arguments condition An expression that when it evaluates to a non-zero value, the loop ends 
execution 

example // repeat until input 1 is on 
repeat until IN#1 { 
 // ... 
} 

The repeat until statement executes a series of statements until a condition evaluates to a non-zero value.  The 

condition is evaluated after the statements execute.   

The statements within the block are at least executed one time. 

Repeat until may not appear in an atomic block, but may be nested. 

 

While 
syntax while condition { statement; … statement; }; 

arguments condition An expression that when it evaluates to a zero value, the loop does not 
execute 

example // execute while  
while IN#1 { 
 wait for rise of 1; 
 wait for fall of 1; 
} 

The while statement executes a series of statements while a condition evaluates to a non-zero value.   

The statements within the block do not get executed if the condition evaluates initially to a zero value.  The 

condition is reevaluated after all statements execute.   

While statements may not appear in an atomic block, but may be nested. 

On (Axes) 
syntax on (axes) { statement; … statement; } 

arguments axes A space-delimited list of numeric axes to execute AML statements on 

example // enable axis 1, 2 and 4 
// then, start the POB named “findHome” on those axes 
on (1 2 4) { 
 enable drive; 
 start findHome FG; 
} 

The on (axes) statement executes a series of statements on one or more axes in a multi-axis environment. 

⚠ This statement may only be used on the supervisory controller. 

⚠ Statements are executed in background (BG). 

 

On Axes 
syntax on axes { axis n1: statement; … statement; axis n2: statement; … statement; … } 

arguments axes A space-delimited list of numeric axes to execute AML statements on 

example // enable axis 1 and start the POB named “index” on axis 2 
on axes { 
    axis 1: 
 enable drive; 
    axis 2: 
 start index FG; 
} 

The on statement executes a series of differing statements on one or more axes in a multi-axis environment. 

⚠ This statement may only be used on the supervisory controller. 

⚠ Statements are executed in background (BG). 

EVENT STATEMENTS 

 

On Fault 
syntax on fault handler 

arguments handler A handler of one the following forms: 
 start name 
 ignore 
where name is the name of a POB 

example // if a fault occurs, start the fault handler named drive_faulted 
on fault start drive_faulted; 
// don’t do anything if a fault occurs 
on fault ignore; 

The on fault statement defines what should occur when a drive fault or run-time programming fault occurs. 

When the handler is specified to start a POB, the POB named name will be started as a BG task when a drive fault 

occurs. 

⚠ When a fault occurs, all other running POBs are first stopped and only then is the new BG task started. 

When the handler is specified as ignore (the default at power-up), no fault handling POB is started when a fault 

occurs. 

MOTION STATEMENTS 

Before defining the available motion related statements in AML, an understanding of the target-generator 

operating states is in order. 

 

TARGET GENERATOR OPERATING STATES 

A diagram of the operating states and transitions between them appears on the following page. 

Stopped 

The stopped state is the initial state of the target-generator.  When in this state, the target-generator is not 

generating new target positions for the position loop to follow, but the position loop is active. 

Positioning 

The positioning state is when the target-generator is calculating a series of target positions based on time, such as 

during a simple incremental move.  This is a temporary state – when the generation is complete, the target-

generator enters the stopped state. 

Following 

The following state is when the target-generator is calculating a series of target positions based on a master 

position, such as occurs during gearing or master/slave operation.  This master position should not be confused 

with the master controller (supervisor) in the first slot of an MX controller. 

CAM 

The CAM state is when the target-generator is calculating a series of target positions using an interpolation table of 

consisting of (position, position) points. 

Position-Based Velocity 

The position-based velocity (PBV) state is when the target-generator is integrating a command velocity producing a 

target position.  This state is commonly used when there is no real end-position, such as during a jogging operation 

or when moving to a hard stop. 

UTILITY MOTION STATEMENTS 

 

Enable (Drive) 
syntax enable drive 

example // Enable the drive 
enable drive; 

The enable drive statement enables the drive and prepares the target-generator for motion generation. 

 

Disable (Drive) 
syntax disable drive; 

example // Disable the drive 
disable drive; 

The disable drive statement disables the power section of the drive.  If motion is in progress upon executing the 

disable, the motion will be abruptly terminated. 

 

Motion Stop 
syntax motion stop 

motion stop rate: rate 

arguments rate An expression that is evaluated as the deceleration rate 

example // Stop motion abruptly 
motion stop; 
// Stop motion, but slew to zero speed 
motion stop rate: 100; 

The motion stop statement terminates any active target generation and enters the stopped state. 

In the first form of the statement, the target-generator simply stops generating a new target position. 

In the second form of the statement, the target-generator slews the speed to zero at the specified rate. 

In this form of the statement, since the target position is not immediately stopped, the resulting ending target 

position may be well beyond where it was before the slew to zero speed started. 

 

Trigger Next 
syntax trigger next 

example // set trigger 
trigger next; 
… 
// ‘UTRIG’ trigger will occur here 
move trap to: 3.75 rate: 1000; 

The trigger next statement arms a UTRIG data-capture trigger that will occur on and at the start of the next motion 

statement. 

Set Target 
syntax set target position: position 

set target counts: counts 
arguments position An expression that is evaluated as the new value for the absolute target 

position in user-units (“uu”) 
counts An expression that is evaluated as the new value for the absolute target 

position (feedback counts) 
example // Zero the target position 

// For example: 
//    TPOS=1.000 and FPOS=1.001 before 
//    TPOS=0.000 and FPOS=0.001 after  
set target position: 0; 
// Set the target position counter to 1000 counts 
// For example: 
//    TPOSC=9000 and FPOSC=9002 before 
//    TPOSC= 100 and FPOSC= 102 after 
set target counts: 100; 

The set target statement changes the value of the absolute target position counter, while maintaining any existing 

following error. 

One such application of this statement is to declare where “zero position” is after the execution of a homing 

procedure is complete. 

Following error is intentionally maintained, as that is what is usually desired. 

 

Offset Target 
syntax offset target position: position 

offset target counts: counts 
arguments position An expression that is evaluated as the value to add to the absolute target 

position in user-units (“uu”) 
counts An expression that is evaluated as the value to add to the absolute target 

position (feedback counts) 
example // Bump the target position by 1.000 

// For example: 
//    TPOS=1.000 and FPOS=1.001 before 
//    TPOS=2.000 and FPOS=2.001 after  
offset target position: 1.0; 

The offset target statement adds (offsets) the value of the absolute target position counter, while maintaining any 

existing following error. 

Following error is intentionally maintained, as that is what is usually desired. 

Zero Following Error 
syntax zero following error 

example // Zero the following error and call this position zero 
zero following error; 
set target position: 0; 

The zero following error statement removes any following error in the position loop by copying the target position 

counter into the feedback position counter. 

For example, the zero following error statement is useful to reduce or eliminate built up torque after homing on a 

hard stop. 

 

Zero Target (Position) 
syntax zero target 

example // The two following statements are the equivalent. 
zero target; 
set target position: 0; 

The zero target position statement is a short form of set target position with an argument of zero. 

 

Couple Master 
syntax couple master source 

arguments source The source for the master position: 
 virtual  an internally generated source 
 hw n  from a slave hardware source #n 
 axist n  from another axis’ target position (multi-axis) 
 axisf n  from another axis’ feedback position (multi-axis) 
 superv n  the supervisory controller’s hardware source #n 
 vsuperv  the supervisory controller’s virtual 
 none  no source 

example // Couple the master to the “virtual encoder” 
couple master virtual; 
// “Move” the “virtual encoder” at 100 counts/sec for 100000 counts 
// ...rate limit the generated feedback to 5000 counts/sec/sec 
move virtual at: 100 for: 100000 rate: 5000; 
// Wait until “virtual encoder” movement is complete 
wait virtual; 
// Couple back to hardware 
couple master hw 1; 

The couple master changes the source of where the master position feedback is derived from. 

The source can be virtual, which is an internally generated source (and controlled by means of move virtual 

statements), physical hardware (hw).  In some following applications, it is desired to be able to decouple the slave 

axis while still actively operating in following mode – the couple master none allows such as it can be used to 

decouple the slaved axis from the (master) source, perform some operation, and then couple it back. 

In a multi-axis environment the source may be the supervisor’s encoder position or virtual position (superv or 

vsuperv) or another axis’ target (axist) or feedback (axisf) position. 

Wait (For In Position) 
syntax wait for in position; 

example // Zero the following error and call this position zero 
zero following error; 
set target position: 0; 

The wait for in position statement waits until the target generator is finished generating new target positions and 

that the feedback position is within a window of that ending target position. 

Program flow continues onward after issuing a motion statement – the POB is free to perform other operations 

(with the exception of issuing a conflicting motion statement) before waiting and checking for the completion of 

the intended motion. 

This statement is applicable only when the axis is operating in the positioning state (time-oriented motion). 

⚠ The wait for in position statement cannot appear within an atomic block. 

 

Move Virtual 
syntax move virtual at: freq for: counts rate: rate 

move virtual at: freq rate: rate 
move virtual stop 

arguments freq An expression that is evaluated as the frequency to generate counts (/sec) 
counts An expression that is evaluated as the total number of counts to generate 
rate An optional expression that is evaluated as an maximum rate of change for 

the frequency count generation: a value of 0 (implied when no rate is given) 
means instantaneous change of frequency 

example // Generate 100000 pulses at 100 counts/sec on the virtual encoder 
move virtual at: 100 for: 100000; 
wait virtual; 
// Generate 100000 pulses at 500 counts/sec with a rate of change of 10000 counts/sec/sec 
move virtual at: 500 for: 100000 rate: 10000; 
wait virtual; 

The first two forms of the move virtual statement begins the generation of virtual encoder pulses and are used in 

conjunction with the couple master virtual statement.  The first form limits the number of generated pulses to a 

specified number of counts. 

⚠ An error is generated if this form of the move virtual statement is issued and a move virtual is in progress. 

The second form of the move virtual statement alters the frequency (and possibly the rate of change to seek this 

new frequency) at which the pulses are generated. 

The third form of the move virtual statement immediately stops the generation of pulses. 

Wait (Virtual) 
syntax wait virtual; 

example // start virtual master pulse generation 
move virtual at: 500 for: 100000 rate: 10000; 
// wait until pulse generation is complete 
wait virtual; 

The wait virtual statement waits until the virtual pulse generator is finished generating new counts. 

 

(Enter) Position Seek Mode 
syntax seek on; 

example // Zero the following error and call this position zero 
zero following error; 
set target position: 0; 
R0 = 5;   // absolute position 
R1 = 10;  // maximum acceleration/deceleration 
R2 = 1000; // maximum velocity 
// Start position seek (and therefore head to absolute position 5.0) 
seek on; 

The seek on statement changes the mode of the target generator where is constantly seeks the absolute position 

specified in the register R0. 

While attempting to seek this position, the target generator will limit its generated velocity to the unsigned value 

contained in the R2 register.  Also, the unsigned value contained in the R1 register is used to limit the 

acceleration/deceleration rate. 

 

(Exit) Position Seek Mode 
syntax seek off; 

example // Stop seeking the position in R0 
seek off; 

The seek off statement changes to mode of the target generator from seeking to the stopped state. 

SIMPLE MOTION STATEMENTS (ABSOLUTE) 

 

Move To 
syntax move to: abspos 

move to: abspos rate: rate 
move to: abspos acc: arate dec: drate 

arguments abspos An expression that is evaluated as the absolute end position 

rate An expression that is evaluated as an unsigned acceleration/deceleration 
rate 

arate An expression that is evaluated as an unsigned acceleration rate 
drate An expression that is evaluated as an unsigned deceleration rate 

example // Move using the default acc/dec rates (no S-curve generation) 
move to: 1.5+x; 
// Another move 
move to: 3.75 rate: 100; 

The move to statement generates a triangular move to an absolute position (abspos) consisting of two segments 

namely an acceleration segment and a deceleration segment. 

If no rates are specified (first form of the statement), then the default acceleration and deceleration rates are 

used. 

 

Move At To 
syntax move at: maxvel to: abspos 

move at: maxvel to: abspos rate: rate 
move at: maxvel to: abspos acc: arate dec: drate 

arguments maxvel An expression that is evaluated as a maximum velocity limit 
abspos An expression that is evaluated as the absolute end position 

rate An expression that is evaluated as an unsigned acceleration/deceleration 
rate 

arate An expression that is evaluated as an unsigned acceleration rate 
drate An expression that is evaluated as an unsigned deceleration rate 

example // Move using the default acc/dec rates (no S-curve generation) 
move at: 10 to: 1.577; 
// Another move 
move at: maxspeed to: 3.75 rate: 100; 

The move at to statement generates a trapezoidal move to an absolute position (abspos) consisting of three 

segments namely an acceleration segment, a constant velocity segment and a deceleration segment.  If the 

specified maximum velocity limit is unattainable with the specified arguments, then a triangular move is 

generated. 

If no rates are specified (first form of the statement), then the default acceleration and deceleration rates are 

used. 

Move Trap To 
syntax move trap to: abspos 

move trap to: abspos rate: rate 
move trap to: abspos acc: arate dec: drate 

arguments abspos An expression that is evaluated as the absolute end position 
rate An expression that is evaluated as an unsigned acceleration/deceleration 

rate 
arate An expression that is evaluated as an unsigned acceleration rate 
drate An expression that is evaluated as an unsigned deceleration rate 

example // Move using the default acc/dec rates 
move trap to: endpos; 
// Non-default acc/dec 
move trap to: 3.75 rate: 1000; 

The move trap to statement generates a trapezoidal move to an absolute position (abspos) consisting of three 

segments namely an acceleration segment, a constant velocity segment and a deceleration segment.  The second 

form generates trapezoidal segments such that each of the segments is of equal time:  of ⅓ the time will be spent 

accelerating, ⅓ of the time will be spent at a constant velocity, and ⅓ of the time will be spent decelerating. 

 

Move In To 
syntax move in: time to: abspos 

arguments time An expression that is evaluated as a time, specified in seconds 

abspos An expression that is evaluated as the absolute end position 
example // Move in 100 milliseconds to 1.5 

move in: 0.100 to: 1.5; 

The move in to statement generates a trapezoidal move to an absolute position (abspos) consisting of three 

segments namely an acceleration segment, a constant velocity segment and a deceleration segment.  The 

calculated acceleration, deceleration and maximum velocity are governed by the time argument that dictates the 

total time the move should take. 

  



 
 
 

34-2201   Allied Motion Programmability Standard | AML 45 

 

Move Alter To 
syntax move alter to: abspos rate: rate 

arguments abspos An expression that is evaluated as the new absolute end position 
rate An expression that is evaluated as the rate to use for both acceleration and 

deceleration 

example // Move to 1.5 in 2.5 seconds 
move in: 2.5 to: 1.5; 
wait 500 ms; 
// Changed my mind – head instead to 2.0 
move alter to: 2.0 rate: 1000; 

The move alter to statement changes the end position on-the-fly for an active time-based move command (either 

absolute or incremental). 

This statement specifies a new absolute end position with a specified acceleration and deceleration rate. 

SIMPLE MOTION STATEMENTS (INCREMENTAL) 

 

Move For 
syntax move for: incpos 

move for: incpos rate: rate 
move for: incpos acc: arate dec: drate 

arguments incpos An expression that is evaluated as the incremental position 

rate An expression that is evaluated as an unsigned acceleration/deceleration 
rate 

arate An expression that is evaluated as an unsigned acceleration rate 
drate An expression that is evaluated as an unsigned deceleration rate 

example // Move 1 using the default acc/dec rates (no S-curve generation) 
move for: 1.0; 
// Another move 
move for: -4 rate: 100; 

The move to statement generates a triangular move for a specified position displacement (incpos) consisting of 

two segments namely an acceleration segment and a deceleration segment. 

If no rates are specified (first form of the statement), then the default acceleration and deceleration rates are 

used. 

 

Move At For 
syntax move at: maxvel for: incpos 

move at: maxvel for: incpos rate: rate 
move at: maxvel for: incpos acc: arate dec: drate 

arguments maxvel An expression that is evaluated as a maximum velocity limit 
incpos An expression that is evaluated as the incremental position 

rate An expression that is evaluated as an unsigned acceleration/deceleration 
rate 

arate An expression that is evaluated as an unsigned acceleration rate 
drate An expression that is evaluated as an unsigned deceleration rate 

example // Move using the default acc/dec rates (no S-curve generation) 
move at: 10 for: 1.5; 
// Another move 
move at: maxspeed for: 3 rate: 100; 

The move at for statement generates a trapezoidal move for a specified position displacement (incpos) consisting 

of three segments namely an acceleration segment, a constant velocity segment and a deceleration segment.  If 

the specified maximum velocity limit is unattainable with the specified arguments, then a triangular move is 

generated. 

If no rates are specified (first form of the statement), then the default acceleration and deceleration rates are 

used. 

Move Trap For 
syntax move trap for: incpos 

move trap for: incpos rate: rate 
move trap for: incpos acc: arate dec: drate 

arguments incpos An expression that is evaluated as the incremental position 
rate An expression that is evaluated as an unsigned acceleration/deceleration 

rate 
arate An expression that is evaluated as an unsigned acceleration rate 
drate An expression that is evaluated as an unsigned deceleration rate 

example // Move using the default acc/dec rates 
move trap for: delta; 
// Non-default acc/dec 
move trap for: 3.5 rate: 500; 

The move trap to statement generates a trapezoidal move for a specified position displacement (incpos) consisting 

of three segments namely an acceleration segment, a constant velocity segment and a deceleration segment.  The 

second form generates trapezoidal segments such that each of the segments is of equal time:  of ⅓ the time will be 

spent accelerating, ⅓ of the time will be spent at a constant velocity, and ⅓ of the time will be spent decelerating. 

 

Move In To 
syntax move in: time for: abspos 

arguments time An expression that is evaluated as a time, specified in seconds 
incpos An expression that is evaluated as the incremental position 

example // Move in 100 milliseconds 1.5 uu 
move in: 0.100 for: 1.5; 

The move in to statement generates a trapezoidal move for a specified position displacement (incpos) consisting of 

three segments namely an acceleration segment, a constant velocity segment and a deceleration segment.  The 

calculated acceleration, deceleration and maximum velocity are governed by the time argument that dictates the 

total time the move should take. 

Move Alter For 
syntax move alter for: incpos rate: rate 

arguments incpos An expression that is evaluated as an incremental position relative to the 
current position 

rate An expression that is evaluated as the rate to use for both acceleration and 
deceleration 

example // Move to 1.5 in 2.5 seconds 
move in: 2.5 to: 1.5; 
wait 500 ms; 
// Changed my mind – stop here quickly 
move alter for: 0 rate: 20000; 

The move alter for statement changes the end position on-the-fly for an active time-based move command (either 

absolute or incremental). 

This statement specifies a relative (to the current position) end position with a specified acceleration and 

deceleration rate. 

POSITION-BASED VELOCITY MOTION STATEMENTS 

 

PBV Enter/Exit 
syntax pbv enter 

pbv exit 
example // Enter PBV state 

pbv enter; 
// ... 
// exit PBV state 
pbv exit; 

The pbv enter statement causes the target generator to enter the position-based velocity (PBV) state (from the 

stopped state). 

⚠ An error is generated if a pbv enter statement is issued when not in the stopped state. 

The pbv exit statement causes the target generator to exit the position-based velocity (PBV) state and enter the 

stopped state.  An error is not generated if the target generator is not in the PBV state. 

 

PBV At 
syntax pbv at: vel 

pbv at: vel rate: rate 
arguments vel An expression that is evaluated as the (new) velocity for use in the 

position-based velocity state 
rate An expression that is evaluated as an unsigned acceleration/deceleration 

rate 
example pbv enter; 

// Input #1 is jog input – wait for rise 
wait for rise of 1; 
// Start jogging at 1 uu/sec 
pbv at: 1.0 rate: 1000; 
wait for fall of 1; 
// Stop jogging 
pbv at: 0 rate: 2000; 
pbv exit; 

The pbv at statement causes the target generator to change the PBV velocity. 

The first form of the statement instantaneously changes the PBV velocity (no regard to rate limiting). 

The second form of the statement rate limits the change in the PBV velocity. 

In the PBV state, target positions are generated based upon the current PBV velocity without regard to an end 

position.  The PBV state is useful for jogging an axis, or during homing procedures where the end position is 

unknown or undefined. 

The current PBV velocity can be altered any number of times by issuing additional pbv at statement. 

PBV Stop 
syntax pbv stop within: displacement 

arguments displacement An expression that is evaluated as an unsigned position displacement to stop 
within 

example // Start moving at 2 uu/sec 
pbv at: 2.0 rate: 1000; 
wait for fall of 2; 
// Stop 1 uu after 
pbv stop within: 1; 

The pbv stop within statement causes the target generator to command a zero velocity (starting from the current 

PBV velocity) at such a deceleration rate that the specified position displacement is consumed before stopping. 

The target generator will automatically sign the displacement appropriately based upon the sign of the current PBV 

velocity. 

If the current PBV velocity is zero, the statement will be effectively ignored. 

COMPLEX MOTION STATEMENTS 

Complex motion statements specify one or more jerk parameters for S-curve generation. 

 

CMove At To 
syntax cmove at: maxvel to: abspos rate: rate jerk: jerk 

cmove at: maxvel to: abspos acc: arate dec: drate jacc: jacc jdec: jdec 

arguments maxvel An expression that is evaluated as an unsigned velocity limit 
abspos An expression that is evaluated as the absolute end position 
rate An expression evaluated as an unsigned acceleration/deceleration rate 
jerk An expression that is evaluated as an unsigned jerk limit 
arate An expression that is evaluated as an unsigned acceleration rate 
drate An expression that is evaluated as an unsigned deceleration rate 
jacc An expression that is evaluated as an unsigned jerk limit used during 

acceleration 
jdec An expression that is evaluated as an unsigned jerk limit used during 

deceleration 

example // Move symmetrical 
cmove at: 10 to: x rate: 1000 jerk: 500; 
// Move unsymmetrical 
cmove at: 15 to: x acc: 1000 dec: 700 jacc: 1500 jdec: 400; 

The cmove at to statement generates a complex move to an absolute position (abspos) consisting of up to 7 

segments. 

The first form of the statement generates a symmetric move, whereas the second form generates an 

unsymmetrical move. 

 

CMove At For 
syntax cmove at: maxvel for: incpos rate: rate jerk: jerk 

cmove at: maxvel for: incpos acc: arate dec: drate jacc: jacc jdec: jdec 
arguments maxvel An expression that is evaluated as an unsigned velocity limit 

incpos An expression that is evaluated as the displacement 
rate An expression evaluated as an unsigned acceleration/deceleration rate 
jerk An expression that is evaluated as an unsigned jerk limit 
arate An expression that is evaluated as an unsigned acceleration rate 
drate An expression that is evaluated as an unsigned deceleration rate 

jacc An expression that is evaluated as an unsigned jerk limit used during 
acceleration 

jdec An expression that is evaluated as an unsigned jerk limit used during 
deceleration 

example // Move symmetrical 
cmove at: 10 for: x rate: 1000 jerk: 500; 
// Move unsymmetrical 
cmove at: 15 for: x acc: 1000 dec: 700 jacc: 1500 jdec: 400; 

The cmove at for statement generates an incremental (relative) complex move consisting of up to 7 segments. 

The first form of the statement generates a symmetric move, whereas the second form generates an 

unsymmetrical move. 

TABLE-BASED MOTION STATEMENTS 

Motion tables are another way to define a trajectory.  AML supports a number of types of motion tables. 

All tables are composed as relative to a starting entry of (0.0, 0.0).  Therefore, when executed, the table is relative 

to the current position of the axis. 

Spline 

The Spline algorithm generates a smooth cubic-spline (3rd-order polynomial) fitted profile from a table consisting of 

a series of (time, position) or (position, position) points.  The profile is generated in such a way both the velocity 

and acceleration is continuous at and between points. 

The algorithm guarantees that the generated curve passes through each of the specified points. 

Linear 

The Linear algorithm does a linear fit of position from a table consisting of a series of (time, position) or (position, 

position) points. 

The algorithm guarantees that the generated curve passes through each of the specified points. 

B-Spline 

The B-Spline algorithm generates a smooth B-spline (a generalized Bézier curve) profile from a table consisting of a 

series of (time, position) or (position, position) points.  The profile is generated in such a way both the velocity and 

acceleration is continuous. 

The algorithm does not guarantee that the generated position passes through each of the specified points – 

however, the final generated position will be equal to the last specified position. 

 

 

⚠ For all point-pairs in the table, (x, y), the ‘x’ entries must be increasing in value throughout the table. 

Table (Clear) 
syntax table clear n 

arguments n The table number 

example // Clear table 1 
table clear 1; 
// ... 

The table clear statement clears all points from a table. 

 

Table (Add) 
syntax table add n (x , y) 

arguments n The table number 

x An expression for the ‘x’ entry of the point pair 

y An expression for the ‘y’ entry of the point pair 

example // Clear table 1, and add a few points 
table clear 1; 
table add 1 (0.1, 1.5); 
table add 1 (0.2, 3.7); 
table add 1 (0.35, 4.1); 
table add 1 (0.5, 5.0); 
// ... 

The table add statement adds a point to the table.  If the table is empty, a (0.0, 0.0) entry is inserted before adding 

this point. 

 

Table (From File) 
syntax table add n from “filename” 

arguments n The table number 

filename The name of the file to load points from 

example // Clear table 1, then add points from a file 
table clear 1; 
table add 1 from file “part.dat” 
// ... 

The table add statement adds points to a table from a file containing a series of IEEE-754 single-precision floats.   

The size of the file (in bytes) must be divisible by 8, or a run-time error will occur. 

The table must be empty or a run-time error will occur. 

The first entry in the file must be the point-pair (0.0, 0.0) or a run-time error will occur. 

Build (Spline) 
syntax build spline n 

 

arguments n The table number 

example // Build a Spline table 
build spline 1; 
// ... 

The build spline statement builds a Spline table for later execution. 

 

Build (Linear) 
syntax build linear n 

 
arguments n The table number 

example // Build a linear table 
build linear 1; 
// ... 

The build linear statement builds a Linear table for later execution. 

 

Build (BSpline) 
syntax build bspline n 

build bspline3 n 
build bspline4 n 
build bspline5 n 
build bspline6 n 

arguments n The table number 

example // Build a degree-3 B-Spline table 
build bspline 1; 
// ... 
// Build a degree-5 B-Spline table 
build bspline5 2; 

The build bspline statement builds a B-Spline table for later execution. 

Table Start 
syntax table start n tscale: timescale pscale: pscale 

arguments n The table number to use for motion 
timescale An expression that will be evaluated and used as a scale factor for the time 

values within the table 
pscale An expression that will be evaluated and used as a scale factor for the 

position values within the table 
example // Run table #2 

table start 2 tscale: 1 pscale: 1; 

The table start statement begins profile generation using the specified motion table containing (time, position) 

pairs. 

 

Table Stop 
syntax table stop 

example // Stop the active running table 
table stop; 

The table stop statement stops (time) table-based target profile generation. 

 

CAM Start 
syntax cam start n mscale: mscale sscale: sscale 

arguments n The table number to use for CAM operation 
mscale An expression that will be evaluated and used as a scale factor for the 

master-feedback position values within the table 
sscale An expression that will be evaluated and used as a scale factor for the 

slave (axis) position values within the table 
example // Run table #3 

cam start 3 mscale: 1 sscale: 1.5; 

The cam start statement begins target profile generation using the specified (CAM) table containing (position, 

position) points.  The CAM operation will continue unless a cam stop is issued or a fault occurs. 

CAM tables are a way to simulate a mechanical CAM.  CAM tables are built and executed in a similar way to their 

time-based counterparts, with the exception that the last position in the CAM table must be zero as CAM tables 

repeat and without the last position equaling the first, a discontinuity would occur. 

 

CAM Stop 
syntax cam stop 

example // Stop the active running CAM table 
cam stop; 

The cam stop statement ends CAM table profile generation and enters the stopped target generator state. 

FOLLOWING MOTION STATEMENTS 

 

Follow Enter / Exit 
syntax follow enter 

follow exit 
example // Enter following mode 

follow enter; 
// ... 
// Exit following mode 
follow exit; 

The follow enter statement enters following mode from a stopped target generator state. 

The follow exit statement exits following mode and enters the stopped target generator state. 

 

Gear At 
syntax gear at: numerator : denominator 

arguments numerator An expression that evaluates to the gear ratio numerator 

denominator An expression that evaluates to the gear ratio denominator 
example // Enter following mode 

follow enter; 
// Gear up at 1:1 
gear at: 1 : 1; 
// ... 

The gear at statement changes the gear ratio.  The change is made instantaneously. 

 

Gear At In 
syntax gear at: numerator : denominator in: icounts 

arguments numerator An expression that evaluates to the gear ratio numerator 
denominator An expression that evaluates to the gear ratio denominator 
icounts An expression that evaluates to the number of master counts to apply the 

gear ratio change within 
example // Enter following mode 

follow enter; 
// Gear up at 1:1 in 1000 master counts 
gear at: 1 : 1 in: 1000; 
// ... 
// Gear up at 0:1 in 1000 master counts 
gear at: 0 : 1 in: 1000; 
// ... 

The gear at in statement changes the gear ratio.  The change is made gradually over some number of master 

counts (icounts). 

Gear At In After 
syntax gear at: numerator : denominator in: icounts after: acounts 

arguments numerator An expression that evaluates to the gear ratio numerator 
denominator An expression that evaluates to the gear ratio denominator 
icounts An expression that evaluates to the number of master counts to apply the 

gear ratio change within 

acounts An expression that evaluates to the delay (in master counts) before the 
change begins to occur 

example // Enter following mode 
follow enter; 
// Gear up at 1:1 in 1000 master counts after 2000 have passed 
gear at: 1 : 1 in: 1000 after: 2000; 
// ... 

The gear at in after statement changes the gear ratio.  The change is made gradually over some number of master 

counts (icounts) and is started after some number of master counts has occurred (acounts). 

 

Gear For In 
syntax gear for: scounts in: mcounts 

arguments scounts An expression that evaluates to counts of the slave 
mcounts An expression that evaluates to counts of the master 

example // Enter following mode 
follow enter; 
// Gear up at 1:1 in 1000 master counts 
gear at: 1 : 1 in: 1000; 
// Consume 1200 slave counts in the next 1000 master counts 
gear for: 1200 in: 1000; 

The gear for in statement calculates and temporarily modifies the active gear ratio such that a certain number of 

slave counts (scounts) are consumed in a certain number of master counts (mcounts). 

The adjustment can be thought of as superimposing a triangular move while gearing is active. 

 

Gear For In After 
syntax gear for: scounts in: mcounts after: acounts 

arguments scounts An expression that evaluates to counts of the slave 

 mcounts An expression that evaluates to counts of the master 
acounts An expression that evaluates to a delay (in master counts) before the 

correction action occurs 
example // Enter following mode 

follow enter; 
// Gear up at 1:1 in 1000 master counts 
gear at: 1 : 1 in: 1000; 
// Consume 1200 slave counts in the next 1000 master counts 
// start the correction after 1100 master encoder counts 
gear for: 1200 in: 1000 after: 1100; 

The gear for in after statement calculates and temporarily modifies the active gear ratio such that a certain 

number of slave counts (scounts) are consumed in a certain number of master counts (mcounts). 

The adjustment, which occurs after a delay of acounts master counts, can be thought of as superimposing a 

triangular move while gearing is active. 

Offset Slave 
syntax offset slave by: scounts in: time 

arguments scounts An expression that evaluates to counts of the slave 
time An expression that evaluates to the time to apply the correction (in 

seconds) 
example // Enter following mode 

follow enter; 
// Gear up at 1:1 in 1000 master counts 
gear at: 1 : 1 in: 1000; 
// Consume an additional 100 slave counts in the next 10 seconds 
offset slave by: 100 in: 10; 

The offset slave statement calculates and temporarily modifies the active gear ratio such that a certain number of 

counts (scounts) of the slave are consumed within a certain time. 

The adjustment can be thought of as superimposing a triangular move while gearing is active. 

 

Wait Slave 
syntax wait slave scounts 

arguments scounts An expression that evaluates to counts of the slave 

example // Enter following mode 
follow enter; 
// Gear up at 1:1 in 1000 master counts 
gear at: 1 : 1 in: 1000; 
// Wait until 1000 slave counts have gone by 
wait slave 1000; 
// deactivate output 1 
set output off (1); 

The wait slave statement waits until a certain number of slave counts (scounts) have occurred. 

 

Wait Master 
syntax wait master mcounts 

arguments mcounts An expression that evaluates to counts of the master 

example // Enter following mode 
follow enter; 
// Gear up at 1:1 in 1000 master counts 
gear at: 1 : 1 in: 1000; 
// Wait until 2000 master counts have gone by 
wait master 2000; 
// activate output 1 
set output on (1); 

The wait master statement waits until a certain number of master counts (mcounts) have occurred. 

Clear Counters 
syntax clear (list) 

arguments list A comma-delimited list of counters to clear (ASCx, AMCx, MSCx, MMCx) 

example // Enter following mode 
follow enter; 
// Clear a few of the auxiliary master counters 
clear master (AMC1, AMC2, AMC4); 
// ... 
// Clear the first modulo slave counter 
clear slave (MSC1); 
// ... 

The clear statement clears one or more auxiliary slave counters (ASCx), modulo slave counters (MSCx), auxiliary 

master counters (AMCx) or modulo master counters (MMCx). 

 

Wait Within 
syntax wait counter counter within lower to upper 

arguments counter One of the slave or master auxiliary or modulo position counters 
lower An expression that evaluates to the lower bounds 
upper An expression that evaluates to the lower bounds 

example // Enter following mode 
follow enter; 
// Gear up at 1:1 in 1000 master counts 
gear at: 1 : 1 in: 1000; 
// wait until input 1 transitions from off to on 
wait for rise of 1; 
// Clear the aux slave counter 1 
clear slave (MSC1); 
// Wait for MSC1 to be within 1000 to 1200 
wait counter MSC1 within 1000 to 1200; 

The wait counter within statement waits until the specified slave or master auxiliary or modulo position counter is 

within some bounds. 

 

Wait Not Within 
syntax wait counter counter not within lower to upper 

arguments counter One of the slave or master auxiliary or modulo position counters 

lower An expression that evaluates to the lower bounds 
upper An expression that evaluates to the lower bounds 

example // Enter following mode 
follow enter; 
// Clear the aux slave counter 1 
clear slave (MSC1); 
// Gear up at 1:1 in 1000 master counts 
gear at: 1 : 1 in: 1000; 
// wait until input 1 transitions from off to on 
wait for rise of 1; 
// Wait for MSC1 to be outside of 1000 to 1200 
wait counter MSC1 not within 1000 to 1200; 

The wait counter outside statement waits until the specified slave or master auxiliary or modulo position counter is 

outside of some bounds. 

Anti Backup (On/Off) 
syntax antibackup state 

arguments state ON/on or OFF/off 

example // Enter following mode 
follow enter; 
// Don’t allow the master to “reverse” 
antibackup on; 

The antibackup statement turns anti-backup on or off. 

 

Wait (Queue) 
syntax wait queue n to var 

arguments n The master-position queue number 
var A variable to store the decremented position to 

example // Wait for an item in the position queue 
wait queue 2 to _delta; 

The wait queue statement waits until a position in a master-position queue has decremented to 0 (or below). 

A master-position queue is an ordered list of positions that decrement in value every time the master-position 

decrements.  Master-position queues can be used, for example, to keep track of upstream sensed objects in a 

labeling application. 

The specified variable, var, is used to hold the final de-queued value pulled from the queue – ideally, it is a zero 

value, but it can be negative meaning that the POB didn’t get a chance to run in time.  The value in var can be used 

to offset some gearing operation. 

 

Move (Slave) 
syntax move slave by: dvar in: time 

move slave by: -dvar in: time 
arguments dvar A variable that holds the displacement to move (in counts) 

time The time to move the slave back within (seconds); must be a numeric constant 
or variable 

example // move the slave back by the value in ASC1 
move slave by: -ASC1 in: 0.2; 

The move slave generates a trapezoidal move for the slave axis.  The move is executed temporarily decoupled from 

the master (in other words, not geared but in the following state). 

The first form uses the dvar variable as is, whereas the second form negates the value from the variable before 

use. 

INPUT EVENT STATEMENTS 

Input Events are used to perform one or more actions when a transition is detected on a digital input.  Input Events 

are managed by digital hardware and therefore can be used for precise operations such as high-speed position 

capture and registration. 

 

Define Event (Input-Triggered) 
syntax define event eventid { 

 transition of n 
 conditions (conditions) 
 actions (actions) 
} 

arguments eventid IEVT1 through IEVT8  
transition Either rise, fall or edge 
n A digital input number 
conditions A comma-delimited list of conditions 
actions A comma-delimited list of actions 

example // Define event IEVT2 to clear AMC2 and start the POB “cap2Detected” 
//   when input #2 rises and input #3 is off 
define event IEVT2 { 
 rise of 2 
 conditions (gate 3 off) 
 actions (clear AMC2, start cap2Detected) 
} 
// enable monitoring of the event 
enable IEVT2; 
// ... 

The define event statement defines (or re-defines if there is an existing definition) what to do when a digital input 

makes a transition.  A list of conditions can govern whether or not the event actions are processed when the 

transition occurs. 

 

CONDITIONS 

 

Gate (Condition) 
syntax gate n state 

arguments n A digital input number 
state Either on or off 

example define event IEVT1 { 
 rise of 1 
 conditions (gate 2 on) 
 // ... 
} 

definition The gate condition is used to “logically and” the transition with the state of another input.  If 
the gating input is not at the specified condition when the IO Event transition occurs, the 
transition will be ignored and no actions will be taken. 

Window (Condition) 
syntax window var min: min max: max 

arguments var A variable used for comparison 

min A constant used as the minimum allowed value 
max A constant used as the maximum allowed value 

example define event IEVT1 { 
 rise of 1 
 conditions (window AMC2 min: 100 max: 250) 
 // ... 
} 

definition The window condition is used to gate the transition with a value of a variable.  If the named 
variable does not lie between the specified bounds when the IO Event transition occurs, the 
transition will be ignored and no actions will be taken. 

 

Outside Window (Condition) 
syntax outside window var min: min max: max 

arguments var A variable used for comparison 
min A constant used as the minimum allowed value 
max A constant used as the maximum allowed value 

example define event IEVT1 { 
 rise of 1 
 conditions (outside window AMC2 min: 100 max: 3900) 
 // ... 
} 

definition The outside window condition is used to gate the transition with a value of a variable.  If the 
named variable lies between the specified bounds when the IO Event transition occurs, the 
transition will be ignored and no actions will be taken. 

 

ACTIONS 

 

Clear Counter (Action) 
syntax clear counter 

arguments counter One of the slave or master auxiliary or modulo 
position counters 

example define event IEVT1 { 
 rise of 1 
 // ... 
 actions (clear MMC1) 
} 

definition The clear counter action clears one of the slave or master auxiliary or modulo position 
counters. 

Enable/Disable IO Event (Action) 
syntax enable eventid 

disable eventid 
arguments eventid IEVT1 through IEVT8 

example define event IEVT1 { 
 rise of 1 
 // ... 
 actions (enable IEVT2, disable IEVT1) 
} 

definition The enable action enables monitoring of an Input Event.  The disable action disables 
monitoring of an Input Event. 

 

Copy (Action) 
syntax copy srcvar destvar 

arguments srcvar The source variable 
destvar The destination variable 

example define event IEVT1 { 
 rise of 1 
 // ... 
 actions (copy AMC1 lastAMC1, clear AMC1) 
} 

definition The copy action stores the value of the source variable (srcvar) to the destination variable 
(destvar). 

 

Start (Action) 
syntax start pob 

arguments pob The name of the POB to start 

example define event IEVT1 { 
 rise of 1 
 // ... 
 actions (start gotCapture) 
} 

definition The start action starts a POB (in foreground). 

 

Queue (Action) 
syntax queue n var 

arguments n The master-position queue to add a value to 
var A variable whose current value will be added to the 

master-position queue 
example define event IEVT1 { 

 rise of 1 
 // ... 
 actions (queue 1 itemOffset) 
} 

definition The queue action adds an item to a master-position queue.  A master-position queue is an 
ordered list of positions that decrement in value every time the master-position decrements.  
Master-position queues can be used, for example, to keep track of upstream sensed objects in 
a labeling application. 

Enable/Disable Event (Input-Triggered) 
syntax enable eventid 

disable eventid 

arguments eventid IEVT1 through IEVT8  

example // enable monitoring of event 3 
enable IEVT3; 
// disable monitoring of event 4 
disable IEVT4; 
// ... 

The enable event statement enables monitoring of an Input Event.  When an event is defined (or redefined), 

monitoring of the event is initially disabled and the enable event statement must be used to enable it. 

The disable event statement disables monitoring of an Input Event.  No action is performed when an Input Event is 

disabled. 

 

INPUT EVENT OBJECTS 

When an event occurs, several fields within the event object are modified.  These fields are accessed using the 

object field operator (‘) on the event object. 

For example, to access the feedback position captured at the time of the event is triggered for event IEV1, the 

expression IEV1’FPOS is used. 

A list of fields within all Input Event objects follows. 

 

Field Description 

FPOS The value of FPOS when the event occurred 

FPOSC The value of FPOSC when the event occurred 

MPOSC The value of the master position counter when the event occurred 

ASCn The value of the auxiliary slave counter #n when the event occurred 

AMCn The value of the auxiliary master counter #n when the event occurred 

MSCn The value of the modulo slave counter #n when the event occurred 

MMCn The value of the modulo master counter #n when the event occurred 

DSCF, DSCT The values of the system variables DSCF or DSCT when the event occurred 

COMMUNICATION STATEMENTS 

 

Unsolicited Message 
syntax umsg port id ( expression ) 

arguments port The UDP port to transmit the message on (1100-9999) 

id An identification code for the message 
expression Data to transmit (transmitted as a float) 

example // Wait for input 1 to rise 
wait for rise of 1; 
// Tell someone about it 
umsg 1200 100 ( TPOS ); 
// ... 

The umsg statement sends an unsolicited message to a waiting listener or listeners.  Receivers can listen on one or 

more ports as well as filtering by id. 

Outgoing messages are buffered locally within the controller for transmission. 

 

Stream 
syntax stream channel var 

stream channel end 
stream end 

arguments channel The channel to stream on (0-15) 

 var The (system global) variable to stream 

example // Stream MPOSC to the world on channel 2 
stream 2 MPOSC; 

The first form of the stream statement streams the value of a system global variable to one or more awaiting 

listeners.  To stream multiple variables, different channels can be used. 

There is no guarantee on the synchronicity or period between transmissions.  If fixed-period, cyclic data is 

required, use the streaming methods and events from the IP class in the ALLNET framework. 

The second form of the stream statement ends streaming on the channel. 

The third form of the stream statement ends all streaming on all channels. 

COMMUNICATION EVENT STATEMENTS 

Communication Events are used to perform one or more actions when a message is received through a fieldbus.  

Unlike IO Events, communication events are buffered on receipt so multiple messages may be queued for later 

processing. 

 

Define Event (Communication Triggered) 
syntax define event eventid { 

 port n 
 conditions (conditions) 
 actions (actions) 
} 

arguments eventid CEVT1 through CEVT8  
n A port number to listen on 
conditions A comma-delimited list of conditions 
actions A comma-delimited list of actions 

example // Define event CEVT1 
define event CEVT2 { 
 port 2 
 conditions (gate 3 off) 
 actions (clear AMC2, start cap2Detected) 
} 
// enable monitoring of the event 
enable IEVT2; 
// ... 

The define event statement defines (or re-defines if there is an existing definition) what to do when a message 

arrives on a channel.  A list of conditions can govern whether or not the event actions are processed when a 

message is received. 

 

CONDITIONS 

 

ID (Condition) 
syntax id n 

arguments n An ID to match 

example define event CEVT1 { 
 port 1 
 conditions (id 100) 
 // ... 
} 

definition The id condition is used to match against a given ID in a message.  Use of multiple id conditions 
is allowed to match several different ids. 

ACTIONS 

 

Enable/Disable Communication Event (Action) 
syntax enable eventid 

disable eventid 
arguments eventid CEVT1 through CEVT8 

example define event CEVT1 { 
 // ... 
 actions (enable CEVT2) 
} 

definition The enable action enables monitoring of a Communication Event.  The disable action disables 
monitoring of a Communication Event. 

 

Copy (Action) 
syntax copy field as type to destvar 

arguments field A field number (index within the message, 0=first) 

 type Either int32 or float32 
destvar The destination variable 

example define event CEVT1 { 
 // ... 
 actions (copy 0 as float32 to index_distance) 
} 

definition The copy action stores the value of a field contained in the message to the destination variable 
(destvar). 

 

Start (Action) 
syntax start pob else drop 

start pob else queue 
arguments pob The name of the POB to start 

example define event CEVT1 { 
 // ... 
 actions (start msgHandler else drop) 
} 

definition The start action starts a POB (in foreground). 
 
If the POB is presently running, and the condition specified is drop then the message if 
effectively ignored. 
 
If the POB is presently running, and the condition specified is queue then the message is 
queued to be processed later when the POB finishes running. 

Enable/Disable Event (Communication Triggered) 
syntax enable eventid 

disable eventid 

arguments eventid CEVT1 through CEVT8  

example // enable monitoring of communication event 3 
enable CEVT3; 
// ... 

The enable event statement enables monitoring of a Communication Event.  When an event is defined (or 

redefined), monitoring of the event is initially disabled and the enable event statement must be used to enable it. 

The disable event statement disables monitoring of the specified event.  No action is performed when an Input 

Event is disabled. 

POB FIELDS 

The following fields are available to query a POB on its status. 

 

Field Description 

‘first Non-zero if this is the first time running this POB since power-up 

‘isRunning Non-zero if the POB is running (or suspended) 

‘isSuspended Non-zero if the POB is suspended 

‘isWaiting Non-zero if the POB is executing any wait statement 

‘notRunning Non-zero if the POB is not running (and not suspended) 

SYSTEM VARIABLES 

 

STATUS VARIABLES 

Symbol Access Scaling/Limits Description 

ANTIB R/O  Non-zero if anti-backup is engaged 

C_NEGONE R/O  A constant negative one (-1.0) 

C_ONE R/O  A constant one (1.0) 

C_PI R/O  A constant pi (3.14159265) 

C_ZERO R/O  A constant zero (0.0) 

ENABLED R/O  Non-zero if the drive is enabled 

FAULT_D R/O  Drive-related fault bit-field 

FAULT_M R/O  Motion-related fault bit-field 

FAULT_N R/O  Non-Motion -related bit-field 

FAULTED R/O  Non-zero if the drive is faulted 

INPOS R/O  Non-zero if motion is complete and the axis is within the “in 
position” window 

INS R/O  A bit-field representing the state of the digital inputs; bits are 
accessible using the ‘#’ operator 

OUTS R/O  A bit-field representing the state of the digital outputs; bits are 
accessible using the ‘#’ operator 

OVER_FE R/O  Non-zero if the following error limit (LIM_FE) has been exceeded 

OVER_N R/O  Non-zero if the negative over-travel input is tripped 

OVER_P R/O  Non-zero if the positive over-travel input is tripped 

OVER_Q R/O  Non-zero if the “quick stop” input is tripped 

OVER_TRQ R/O  Non-zero if a torque limit is active (commanded torque exceed 
LIM_TRQ) 

POB_BG R/O  The number of active background POBs 

POB_FG R/O  The number of active foreground POBs 

RC R/O  Iteration counter 

 

TIMER VARIABLES 

Symbol Access Scaling/Limits Description 

TIME_G R/W seconds Global writeable timer (0 at power-up) 

TIME_I R/W ticks Private, writeable POB-based counter (0 when the POB starts 
execution, incremented by one each time the POB runs) 

TIME_P R/W seconds Private, writeable POB-based timer (0 when the POB starts 
execution) 

TARGET GENERATOR VARIABLES 

Symbol Access Scaling/Limits Description 

ACC R/W user-units/sec2 The default acceleration rate for move statements where none is 
specified 

DEC R/W user-units/sec2 The default deceleration rate for move statements where none 
is specified 

FCPR R/O counts Feedback (slave) counts/motor revolution; a derived value based 
upon the motor feedback source 

FPOS R/O user-units Feedback position 

FPOSC R/O counts Feedback position 

GVRATE R/W  Global VRATE applied to all axes; see VRATE 

LIM_FE R/W user-units Absolute following error limit (0=no following error limit) 

LIM_NOT R/W user-units Negative, software-based over-travel limit 

LIM_POT R/W user-units Positive, software-based over-travel limit 

LIM_TRQ R/W Nm Absolute torque limit 

STATE_TBL R/O  Non-zero if the axis is executing a table start or cam start 
statement 

STATE_FOLLOW R/O  Non-zero if the axis is in the FOLLOWING state 

STATE_PBV R/O  Non-zero if the axis is in the PBV state 

STATE_STOP R/O  Non-zero if the axis is in the STOPPED state 

TPOS R/O user-units Target position 

TPOS_LAST R/O user-units Previous value of TPOS prior to the execution of a set target, 
offset target or zero following error statement 

TPOS_NZ R/O user-units The absolute position of the next index pulse (incremental 
encoder), zero position (absolute encoder or resolver) in the 
negative direction; used in homing 

TPOS_PZ R/O user-units The absolute position of the next index pulse (incremental 
encoder), zero position (absolute encoder or resolver) in the 
positive direction; used in homing 

TPOSC R/O counts Target position 

UUD R/W  User-units scale factor; command arguments are divided by this 
value (after multiplying by UUN) to scale user-units to motor 
revolutions 

UUN R/W  User-units scale factor; command arguments are multiplied by 
this value (and then divided by UUD) to scale user-units to motor 
revolutions 

VRATE R/W  The velocity scale for time-based move statements.  Default 
value of 1.0 at power-up. When set to a value less than 1, 
velocities are scaled down.  When set to a value greater than 1, 
velocities are scaled up. 

ZPULSE R/O  Non-zero if a transition of the marker pulse has occurred on the 
feedback encoder or if absolute feedback is employed 

TARGET GENERATOR VARIABLES (FOLLOWING) 

Symbol Access Scaling/Limits Description 

AMC0-AMC3 R/O counts Auxiliary master counters 

ASC0-ASC3 R/O counts Auxiliary slave counters; based on feedback 

DSCF R/W counts A slave-based counter that decrements for each count of the 
slave feedback 

DSCT R/W counts A slave-based counter that decrements for each count of the 
slave target 

GAIM R/O counts Holds the number of counts of the master that occurred during 
the execution of the last gear at in or gear at in after statement 

GAIS R/O counts Holds the number of counts of the slave that occurred during the 
execution of the last gear at in or gear at in after statement 

GFIM R/O counts Holds the number of counts of the master that occurred during 
the execution of the last gear for in or gear for in after statement 

GFIS R/O counts Holds the number of counts of the slave that occurred during the 
execution of the last gear for in or gear for in after statement 

GR R/O  Active Gear Ratio 

INVM R/W  Set to ‘1’ to invert the direction in which the master feedback 
source counts 

MCPR R/W counts Feedback (master) counts/revolution 

MMC0-MMC3 R/O counts Modulo master counters 

MODM R/W counts Modulus for the master modulo counters; a value of zero 
disables the modulo functionality 

MODS R/W counts Modulus for the slave modulo counters; a value of zero disables 
the modulo functionality 

MPOSC R/O counts Master position 

MSC0-MSC3 R/O counts Modulo slave counters; based upon feedback 

OFFS R/O counts Holds the number of counts of the slave that occurred during 
execution of the last offset slave statement 

SMARK R/O counts A (target-based) slave modulo counter computed on an input 
transition (specified by SMARK_RISE and/or SMARK_FALL); 
SMARK is computed as: SMARK = SPHASE - SMODC 

SMARK_RISE R/W  A bit-field on when to compute SMARK based upon the rising 
transition of one or more inputs 

SMARK_FALL R/W  A bit-field on when to compute SMARK based upon the falling 
transition of one or more inputs 

SMODC R/W counts A modulo slave counter; based upon target.  It is used to 
compute SMARK. 

SPHASE R/W counts The phase modulo slave position used in computing SMARK 

TMC1-TMC2 R/W counts Temporary master counters 

TSC1-TSC2 R/W counts Temporary slave counters; based upon target 

TSC1_RISE R/W  A bit-field on when to clear TSC1 based upon the rising transition 
of one or more inputs 

TSC1_FALL R/W  A bit-field on when to clear TSC1 based upon the falling transition 
of one or more inputs 

TSC2_RISE R/W  A bit-field on when to clear TSC2 based upon the rising transition 
of one or more inputs 

TSC2_FALL R/W  A bit-field on when to clear TSC2 based upon the falling transition 
of one or more inputs 

WAITM R/O counts Holds the actual number of counts of the master that occurred 
during the execution of the last wait slave statement 

WAITS R/O counts Holds the actual number of counts of the slave that occurred 
during the execution of the last wait slave statement 

COMMUNICATION AND COMPILER SUPPORT 

The ALLNET .NET assembly provides the means to communicate with a single-axis drive or multi-axis controller. 

The AMLCompiler .NET assembly provides a means (through the class AMLCompiler) of translating the textual AML 

language into opcode POBs that the VM running on a drive or controller can execute. 

The abstract class IALLNET within the ALLNET assembly provides support for serial (through the subclass 

ALLNET.Serial) and Ethernet (through the subclass ALLNET.IP) communications. 

Although there are many methods available within the AMLCompiler, IALLNET and IP classes, only those methods 

related to programmability are highlighted here. 

 

AMLCOMPILER.AMLCOMPILER CLASS 

AMLCompileResult CompileAML(string code) 

This method translates some AML textual code into a downloadable POB and returns an object (of type 

AMLCompileResult) containing the results of the translation.  The AMLCompileResult class has properties to 

interrogate whether the translation succeeded or not (test if the property, Errors.Count > 0) and a list (the 

property, List<ParseError> Errors) of any errors. 

 

ALLNET.IP CLASS 

bool POBPrepare() 

This method attempts to clear all in-memory (and FLASH file-system) POBs in preparation for subsequent 

UploadPOBFile invocations.  The POBPrepare method stops all running POBs on the drive or controller. 

bool POBControl(POBOPERATION op, int axis = -1, string POBname = null) 

This method starts (either FG or BG), stops, suspends and unsuspends POBs. 

bool UploadPOBFile(string filename, byte[] contents, bool load) 

This method attempts to download the translated POB to the drive.  If the drive has a POB existing with the 

specified filename it is replaced.  The parameter load should be true to load this POB into memory ready for 

execution.  The byte array contents is from the AMLCompileResult method POBasBytes(). 

VIRTUAL MACHINE 

The virtual machine (VM) that executes the variable-length opcodes output by the translator has a stack-based 

architecture optimized for the AML. 

The second generation of the VM (called VMr2) is used in conjunction with AML. 

The VM has 16 general purpose read/write registers (R0-R15 per axis), 16 read-only (within the VM) parameter 

registers (P0-P15) and an integer T (iteration) per POB register. 

 

VM REGISTERS 

Per axis registers R0-R15 can hold integer and floating-point values.  These registers are cleared at power-up.  R0-

R15 are read/write within the VM as well as the outside world via serial (ALLNET/Serial) and Ethernet (ALLNET/IP) 

communications. 

Per controller registers P0-P15 hold floating-point values.  These registers are non-volatile.  These parameter 

registers are read/write from the outside world, using by using serial (ALLNET/Serial) and Ethernet (ALLNET/IP).   

The VM can only read these registers. 

The 16-bit T register (per POB) is exclusively used to count iterations (loops) in a POB.  It is read/write within in a 

POB. 

0X001X CALLX OFFSET 

The CALLX op does a call to a firmware-based function with a variable number of arguments (where x is the 

number of arguments to the call). 

0X0050 MSTPI 

The MSTPI (motion stop immediate) immediately stops target generation and effectively aborts the commanded 

motion in progress. 

0X0051 MSTPS 

The MSTPS (motion stop slewed) stops target generation by ramping the target velocity to 0 using the rate 

programmed in STOPRATE (VMr1) or on the stack (VMr2) effectively ending the commanded motion in progress. 

0x0100 PUSHCI int32 

0x0101 PUSHCL int64 

0x0103 PUSHCD float64 

The PUSHCx ops push numeric constants onto the VM stack. 

0X0106 PUSHX FLAGS, VAR 

The op PUSHX pushes the value of a variable or parameter onto the VM stack.  The parameter var points to a 4-

byte sequence (0x0, 0x0, 0x0, 0x0) followed by the variable name (as a null-terminated string) in the data region, 

unless LCL=1 or SYS=1.  Flags are a bit-field (pertaining to var) of: 

 Bit 0: Reserved (0) 

 Bit 1: 0 = User Variable (UV), 1 = System Variable/Parameter (SYS) 

 Bit 2: 0 = Global Variable (G), 1 = Local (to POB) Variable (LCL) 

0X0107 STORX FLAGS, VAR 

The op STORX pops the value on the stack and stores it into a variable or parameter.  The parameter var points to a 

4-byte sequence (0x0, 0x0, 0x0, 0x0) followed by the variable name (as a null-terminated string) in the data region, 

unless LCL=1 or SYS=1.  Flags are a bit-field (pertaining to var) of: 

 Bit 0: Reserved (0) 

 Bit 1: 0 = User Variable (UV), 1 = System Variable/Parameter (SYS) 

 Bit 2: 0 = Global Variable (G), 1 = Local (to POB) Variable (LCL) 

0X0110 ATBEG 

This op enters atomic mode and is used to implement the atomic block. 

0X0111 ATEND 

This op exits atomic mode and is used to implement the atomic block. 

0X0120 AXIS AXES 

This op defines the axes to forward the following instructions to.  The bit-field axes is used to determine which 

axes to send to. 

0X0121 AXEND 

This op marks the end of forwarding and is used in conjunction with AXIS. 

0X0200 CMPEQ 

0X0201 CMPNE 

0X0202 CMPLT 

0X0203 CMPGT 

0X0204 CMPLE 

0X0205 CMPGE 

These ops compare two arguments on the VM stack and replace those arguments with a 1 on the stack if the 

comparison evaluates to true, otherwise the arguments are replaced with a 0. 

Note: these operations pop b from the stack, then pop a from the stack and leave a==b, a!=b, a<b, a>b, a<=b or 

a>=b on the stack, respectively. 

0X0206 LOR 

0X0207 LAND 

0X0208 LNOT 

These ops logically or, and or invert arguments on the VM stack.  These ops implement the AML operators || (or), 

&& (and) and ! (not). 

0X0209 BOR 

0X020A BAND 

0X020B BNOT 

These ops bitwise or, and or invert arguments on the VM stack.  These ops implement the AML operators | (or), & 

(and) and ~ (not). 

0X020C ADD 

0X020D SUB 

0X020E MUL 

0X020F DIV 

These ops perform calculations of arguments on the VM stack.  These ops implement the AML operators + (add), - 

(subtract), * (multiply) and / (divide). 

Note: these operations pop b from the stack, then pop a from the stack and leave a+b, a-b, a*b or a/b on the 

stack, respectively. 

0X0210 NEG 

This op negates the argument on the VM stack and implements the - (unary negate) AML operator. 

0X0211 MOD 

This op performs a modulo of the arguments on the VM stack and implements the % (modulo) AML operator. 

Note: these operation pops b from the stack, then pops a from the stack and leaves a%b on the stack. 

0X0212 BIT 

This op implements the # (bit-access) AML operator. 

0X0213 ABS 

This op performs an absolute value of the argument on the VM stack. 

0X0225 TINPOS 

This op pushes the value of the system variable INPOS onto the VM stack. 

0X0240 VEC VEC, STORE 

This op implements vector subscripting of the vector numbered vec.  Index is on the top of the VM stack.  When 

store is ‘1’, a store should occur otherwise a load. 

0X0241 TBL TBL, STORE 

This op implements table subscripting of the table numbered tbl.  Indices are on the top of the VM stack.  When 

store is ‘1’, a store should occur otherwise a load. 

0X0270 PS ON 

This op implements the seek on AML statement. 

0X0271 PS OFF 

This op implements the seek off AML statement. 

0X0272 RESET 

This op attempts to reset a hardware or run-time fault. 

0X0273 RET 

This op returns from a BSUB, BSUBZ or BSUBNZ user call. 

0X0280 ALVEC VEC 

This op implements vector memory allocation of the vector numbered vec.  The new size of the vector is on the VM 

stack. 

0X0281 ALTBL TBL 

This op implements table memory allocation of the table numbered tbl.  The new size (number of rows, number of 

columns) is on the VM stack. 

0X0300 BR ADDR 

This op unconditionally alters POB program flow to the specified address. 

0X0301 BRZ ADDR 

This op conditionally alters POB program flow to the specified address when the argument on the VM stack has a 

zero value. 

0X0302 BRNZ ADDR 

This op conditionally alters POB program flow to the specified address when the argument on the VM stack has a 

non-zero value. 

0X0311 BTZ ADDR 

This op conditionally alters POB program flow to the specified address when the “T-register” has a zero value. 

0X0312 BTNZ ADDR 

This op conditionally alters POB program flow to the specified address when the “T-register” has a non-zero value. 

0X0320 BSUB ADDR 

This op unconditionally alters POB program flow and calls the subroutine located at specified address.  The 

subroutine must end with a RET instruction.  Subroutine return addresses are located on a separate return address 

stack within the VM. 

0X0321 BSUBZ ADDR 

This op conditionally alters POB program flow and calls the subroutine located at the specified address when the 

argument on the VM stack has a zero value.  The subroutine must end with a RET instruction.  Subroutine return 

addresses are located on a separate return address stack within the VM. 

0X0322 BSUBNZ ADDR 

This legacy op conditionally alters POB program flow and calls the subroutine at the specified address when the 

argument on the VM stack has a non-zero value.  The subroutine must end with a RET instruction.  Subroutine 

return addresses are located on a separate return address stack within the VM. 

0X0303 WRISE N 

This op waits for the rise (off-to-on transition) of the digital input numbered n. 

0X0304 WFALL N 

This op waits for the fall (on-to-off transition) of the digital input numbered n. 

0X0306 WINPOS 

This op implements the wait for in position statement. 

0X0307 WAITQ N, FLAGS, VAR 

This op implements the wait queue statement.  The value n is the master-position queue number.  The parameter 

var points to a 4-byte sequence (0x0, 0x0, 0x0, 0x0) followed by the variable name (as a null-terminated string) in 

the data region, unless LCL=1 or SYS=1.  Flags are a bit-field (pertaining to var) of: 

 Bit 0: Reserved (0) 

 Bit 1: 0 = User Variable (UV), 1 = System Variable/Parameter (SYS) 

 Bit 2: 0 = Global Variable (G), 1 = Local (to POB) Variable (LCL) 

0X0400 START POB, FLAGS 

This op starts or restarts a POB.  The parameter pob points to a 4-byte sequence (0x0, 0x0, 0x0, 0x1) followed by a 

POB name (as a null-terminated string) in the data region.  Flags are a bit-field (pertaining to pob) of: 

 Bit 0: Reserved (0) 

 Bit 1: 0 = FG, 1 = BG 

 Bit 2: 0 = start (ST), 1 = restart (RST) 

0X0401 END 

This op ends execution of this POB. 

0X0402 ABSUS POB, FLAGS 

This op stops, suspends or reactivates a suspended POB.  The parameter pob points to a 4-byte sequence (0x0, 

0x0, 0x0, 0x1) followed by a POB name (as a null-terminated string) in the data region.  Flags are a bit-field 

(pertaining to pob) of: 

Bit 0: Reserved (0) 

Bit 1: 0 = one (ONE), 1 = all POBs (ALL) 

Bit 2: 0 = stop (STP), 1 = suspend operation (SOP) 

Bit 3: 0 = suspend (SUS), 1 = un-suspend (UNS) 

0X0403 MSTOP FLAGS 

This op stops motion (implements the motion stop statement).  Flags are a bit-field of: 

Bit 0: 0 = abrupt (AB), 1 = rate-limited (RL) 

0X0404 DREN 

This op implements the enable drive statement. 

0X0405 DRDIS 

This op implements the disable drive statement. 

0X0407 DELAY 

This op implements the wait statement – the time (in milliseconds) to wait is on the VM stack. 

0X0408 ZERO 

This op zeros the target position and is equivalent to set target position: 0. 

0X0409 RELAX 

This op zeros the following error. 

0X040A SETT FLAGS 

This op implements the set target and offset target statements.  The argument of the operation is on the VM stack. 

Flags are a bit-field of: 

Bit 0: 0 = set (SET), 1 = offset (OFS) 

Bit 1: 0 = user-units (UU), 1 = counts (CNT) 

0X040B OUT ONBITS, OFFBITS 

This op changes the state of one or more digital outputs.  The bit-field onbits contains which digital outputs to 

activate.  The bit-field offbits contains which digital output to deactivate. 

0X040C SETOUT ONBITS 

This legacy op changes the state of digital outputs.  The bit-field onbits contains which digital outputs to activate. 

0X040D CLROUT OFFBITS 

This legacy op changes the state of digital outputs.  The bit-field offbits contains which digital output to deactivate. 

0X040E PULSEOUT N 

This op activates a digital output for a period of time.  The output number n is part of the instruction, whereas the 

period is on the VM stack. 

  



 
 
 

34-2201   Allied Motion Programmability Standard | Virtual Machine 85 

 

0X040F PLS FLAGS, V, N, PCOUNT, CPAIRS 

This op defines (or redefines) a PLS for the output numbered n. 

The parameter v (and its associated flags) define the reference variable for the PLS.  The parameter v points to a 4-

byte sequence (0x0, 0x0, 0x0, 0x0) followed by the variable name (as a null-terminated string) in the data region, 

unless SYS=1. 

The list of ranged on condition pairs (cpairs) for the PLS immediately follows the op.  The number of on condition 

pairs is contained in pcount. 

Flags are a bit-field (pertaining to v) of: 

 Bit 0: Reserved (0) 

 Bit 1: 0 = User Variable (UV), 1 = System Variable/Parameter (SYS) 

0X0410 PLSON N, FLAGS 

This op enables (or disables) the defined PLS for the output numbered n.  Flags are a bit-field of: 

 Bit 0: 0 = disable PLS (D), 1 = enable PLS (E) 

0X0420 TIMBEG ADDR 

This op begins a timeout block and is used to implement the within block.  The timeout is located on the VM stack.  

The address argument addr is the destination for when the timeout occurs. 

0X0421 TIMEND 

This op ends a timeout block and is used to implement the within block. 

0X0422 POBSTAT POB, N 

This op pushes the state of a POB related status variable (n) on the stack.  The parameter pob points to a 4-byte 

sequence (0x0, 0x0, 0x0, 0x1) followed by a POB name (as a null-terminated string) in the data region. 

0X0430 LDT X 

This op loads the “T-register” with the value x. 

0X0431 INCT 

This op increments the “T-register”. 

0X0432 DECT 

This op decrements the “T-register”. 

0X0433 PUSHT 

This op pushes the “T-register” to the stack. 

0X0434 POPT 

This op pops a value from the stack and stores it in the “T-register”. 

0X0435 TRIGN 

This op implements “trigger next”. 

0X0504 ONFAULT POB, FLAGS 

0X0504 ONFAULT IGNORE 

This op defines the global fault handler.  Flags are a bit-field (pertaining to pob) of: 

 Bit 0: Reserved (0) 

 Bit 1: always 1 

 (Bit 2: bit is set to 1 with IGNORE) 

The parameter pob points to a 4-byte sequence (0x0, 0x0, 0x0, 0x1) followed by a POB name (as a null-terminated 

string) in the data region. 

0X0600 STACK N 

This op allocates n words of VM stack space for use by local POB variables.  The STACK instruction is always the first 

op in a POB (located at address 0). 

0X0602 DRAD ADDR 

This op informs the VM of the location (specified by addr) of the data region.  The DRAD instruction is always the 

second op in a POB (located at address 4). 

0X0603 DATA ENDADDR 

This op marks the data region.  The next code area (if any) is specified in endaddr. 

0X0610 LINK DVAR, FLAGS1, VAR1, CON1, FLAGS2, VAR2, CON2, OFFSET, MIN, MAX 

This op implements the link statement. 

DVAR specifies which destination system variable to link to. 

FLAGS1 are a bit-field (pertaining to VAR1) of: 

 Bit 0: Reserved (0) 

 Bit 1: 1 = System Variable/Parameter (SYS) 

 Bit 2: always 0 

FLAGS2 are a bit-field (pertaining to VAR2) of: 

 Bit 0: Reserved (0) 

 Bit 1: 1 = System Variable/Parameter (SYS) 

 Bit 2: always 0 

The constants CON1, CON2, OFFSET, MIN and MAX are floating-point constants. 

The parameters VAR1 and VAR2 point to a 4-byte sequence (0x0, 0x0, 0x0, 0x0) followed by the variable name (as 

a null-terminated string) in the data region, unless SYS=1.   

0X0611 UNLINK DVAR 

This op implements the unlink statement. 

DVAR specifies which destination system variable to be unlinked. 

0X0612 CAPTURE 

This op implements the capture statement. 

0X0613 CAPABORT 

This op implements the capture abort statement. 

0X0614 UMSG PORT, ID 

This op implements the AML umsg statement.  The result of the expression specified in the umsg statement is on 

the VM stack. 

0X0615 TBLOP TABLE, COMMAND 

This op implements various table-based motion statements. 

 

A command value of 0, indicates a table clear. 

A command value of 1, indicates a table add, where the point-pair is on the VM stack. 

A command value of 4, indicates a build spline. 

A command value of 5, indicates a build linear. 

Command values of 6-9, indicates a build bspline (6=degree 3 through 9=degree 6). 

A command value of 10, indicates a table start.  The arguments to table start are on the VM stack. 

A command value of 11, indicates a table stop. 

A command value of 12, indicates a cam start.  The arguments to cam start are on the VM stack. 

A command value of 13, indicates a cam stop. 

0X0800 CLRC ASC, AMC, MSC, MMC 

This op clears one or more of the following mode counters.  The arguments to the op are bit-fields on which 

counters to clear. 

0X0801 WCW N, FLAGS 

This op implements the wait counter within and wait counter not within statements.  N represents the counter 

number to wait on.  Flags are a bit-field of: 

 Bit 0: 0 = within (WIN), 1 = not within (NWIN) 

 Bit 1: 1 = counter group ASCx (ASC) 

 Bit 2: 1 = counter group AMCx (AMC) 

 Bit 3: 1 = counter group MSCx (MSC) 

 Bit 4: 1 = counter group MMCx (MMC) 

The lower and upper bounds are on the VM stack. 

0X0802 ANTIB N 

This op turns anti-backup on (n=1) or off (n=0). 

0X0803 COUPLE SRC,N 

This op implements couple master. 

0X0804 WAITV 

This op implements wait virtual. 

0X0805 MV_START 

This op implements move virtual which starts a new virtual move.  The first pop from the stack is the rate:.  The 

next pop from the stack is the number of counts (for:).  The last pop from the stack is the frequency (at:). 

0X0806 MV_ALTER 

This op implements move virtual which alters an in-progress virtual move.  The first pop from the stack is the new 

rate:.  The last pop from the stack is the new frequency (at:). 

0X0807 MV_STOP 

This op implements move virtual stop which stops an in-progress virtual move. 

0X0808 F_ENTER 

This op implements follow enter. 

0X0809 F_EXIT 

This op implements follow exit. 

0X080A GEARAT 

This op implements gear at. 

0X080B GEARFOR 

This op implements gear for. 

0X080C WAITS 

This op implements wait slave. 

0X080D WAITM 

This op implements wait master. 

0X080E OFFSLV 

This op implements offset slave. 

0X0900 IEVTDEF N, TRANS, DIN 

This op begins the definition of an input-triggered event.  N is the event ID (1=IEVT1 through 8=IEVT8), trans is 

RISE(1), FALL(2) or EDGE(3), and din is the digital input associated with the event. 

Condition ops (0x0910-0x092F) and/or Action ops (0x0930-0x094F) follow until an IEVTEND op. 

0X0901 IEVTEND 

This op ends the definition of an input-triggered event. 

0X0902 IEVTE N 

This op enables the input-triggered event numbered n. 

0X0903 IEVTD N 

This op disables the input-triggered event numbered n. 

0X0910 IEVTGATE DIN, STATE 

This input-triggered event condition op defines a gating condition.  The din argument is the digital input number to 

gate on and state is ON(1) or OFF(0). 

0X0911 IEVTWIND FLAGS, VAR, MIN, MAX 

This input-triggered event condition op defines a window inside condition on the variable var.  Flags are a bit-field 

(pertaining to var) of: 

 Bit 0: Reserved (0) 

 Bit 1: 0 = User Variable (UV), 1 = System Variable/Parameter (SYS) 

 Bit 2: always 0 

 Bit 8: always 0 

The parameter var points to a 4-byte sequence (0x0, 0x0, 0x0, 0x0) followed by the variable name (as a null-

terminated string) in the data region, unless SYS=1. 

0X0912 IEVTWIND FLAGS, VAR, MIN, MAX 

This input-triggered event condition op defines a window outside condition on the variable var.  Flags are a bit-

field (pertaining to var) of: 

 Bit 0: Reserved (0) 

 Bit 1: 0 = User Variable (UV), 1 = System Variable/Parameter (SYS) 

 Bit 2: always 0 

 Bit 8: always 0 

The parameter var points to a 4-byte sequence (0x0, 0x0, 0x0, 0x0) followed by the variable name (as a null-

terminated string) in the data region, unless SYS=1.   

0X0930 IEVTCC N, FLAGS 

This input-triggered event action op implements clear counter.  N represents the counter number to clear.  Flags 

are a bit-field of: 

 Bit 0: always 0 

 Bit 1: 1 = counter group ASCx (ASC) 

 Bit 2: 1 = counter group AMCx (AMC) 

 Bit 3: 1 = counter group MSCx (MSC) 

 Bit 4: 1 = counter group MMCx (MMC) 

0X0931 IEVTED N, FLAGS 

This input-triggered event action op implements enable/disable event for input events.  N represents the event ID 

to alter.  Flags are a bit-field of: 

 Bit 0: 0 = disable event (D), 1 = enable event (E) 

0X0932 IEVTCPY SRCFLAGS, SRC, DESTFLAGS, DEST 

This input-triggered event action op implements copy.  Both srcflags and destflags are a bit-field (pertaining to src 

and dest, respectively) of: 

 Bit 0: Reserved (0) 

 Bit 1: 0 = User Variable (UV), 1 = System Variable/Parameter (SYS) 

 Bit 2: always 0 

The parameters src and dest points to a 4-byte sequence (0x0, 0x0, 0x0, 0x0) followed by the variable name (as a 

null-terminated string) in the data region, unless SYS=1.   

0X0933 IEVTST POB, FLAGS 

This input-triggered event action op implements start.  Flags are a bit-field (pertaining to pob) of: 

 Bit 0: Reserved (0) 

 Bit 1: always 0 

 Bit 2: always 0 

The parameter pob points to a 4-byte sequence (0x0, 0x0, 0x0, 0x1) followed by a POB name (as a null-terminated 

string) in the data region. 

0X0934 IEVTQ FLAGS, VAR, N 

This input-triggered event action op implements queue.  Flags are a bit-field (pertaining to var) of: 

 Bit 0: Reserved (0) 

 Bit 1: 0 = User Variable (UV), 1 = System Variable/Parameter (SYS) 

 Bit 2: always 0 

The parameter var points to a 4-byte sequence (0x0, 0x0, 0x0, 0x0) followed by the variable name (as a null-

terminated string) in the data region, unless SYS=1.   

  



 
 
 

34-2201   Allied Motion Programmability Standard | Virtual Machine 96 

 

0X0A00 CEVTDEF N, PORT 

This op begins the definition of a communication-triggered event.  N is the event ID (1=CEVT1 through 8=CEVT8) 

and port is the port number associated with the event. 

Condition ops (0x0A10-0x0A2F) and/or Action ops (0x0A30-0x0A4F) follow until a CEVTEND op. 

0X0A01 CEVTEND 

This op ends the definition of an communication-triggered event. 

0X0A02 CEVTE N 

This op enables the communication-triggered event numbered n. 

0X0A03 CEVTD N 

This op disables the communication-triggered event numbered n. 

0X0A10 CEVTID N, ID1, ID2, …, IDN 

This communication-triggered event op implements the id condition.  The n argument specifies the number of 

matching ids to follow. 

0X0A30 CEVTED N, FLAGS 

This communication-triggered event action op implements enable/disable event for communication events.  N 

represents the event ID to alter.  Flags are a bit-field of: 

 Bit 0: 0 = disable event (D), 1 = enable event (E) 

0X0A31 CEVTCPY SRC, FLAGS, DEST, DESTFLAGS 

This communication-triggered event action op implements copy.  Src represents the field number index within the 

message. 

Flags is a bit-field (pertaining to dest) of: 

 Bit 0: 1 = type is int32 

 Bit 1: 1 = type is float32 

Destflags is a bit-field (pertaining to dest) of: 

 Bit 0: Reserved (0) 

 Bit 1: 0 = User Variable (UV), 1 = System Variable/Parameter (SYS) 

 Bit 2: always 0 

The parameter dest points to a 4-byte sequence (0x0, 0x0, 0x0, 0x0) followed by the variable name (as a null-

terminated string) in the data region, unless SYS=1.   

0X0A32 CEVTST QFLAGS, POB, FLAGS 

This communication-triggered event action op implements start. 

Qflags are a bit-field of: 

 Bit 0: 1 = drop 

 Bit 1: 1 = queue 

Flags are a bit-field (pertaining to pob) of: 

 Bit 0: Reserved (0) 

 Bit 1: always 0 

 Bit 2: always 0 

The parameter pob points to a 4-byte sequence (0x0, 0x0, 0x0, 0x1) followed by a POB name (as a null-terminated 

string) in the data region. 

CALLX OFFSET TABLE 

 

0x0000 move to: 

0x0001 move at: to: 

0x0002 move to: rate: 

0x0003 move at: to: rate: 

0x0004 move to: acc: dec: 

0x0005 move at: to: acc: dec: 

0x0010 move for: 

0x0011 move at: for: 
0x0012 move for: rate: 

0x0013 move at: for: rate: 

0x0014 move for: acc: dec: 

0x0015 move at: for: acc: dec: 

0x0040 move trap to: 

0x0042 move trap to: rate: 

0x0044 move trap to: acc: dec: 

0x0050 move trap for: 

0x0052 move trap for: rate: 

0x0054 move trap for: acc: dec: 

0x0060 move trap to: in: 

0x0070 move trap for: in: 

0x0080 move alter to: rate: 

0x0090 move alter for: rate: 

0x00A0 cmove (unsymmetric, to:) 

0x00A1 cmove (symmetric, to:) 

0x00B0 cmove (unsymmetric, for:) 

0x00B1 cmove (symmetric, for:) 

0x00C0 pbv enter 
0x00C1 pbv exit 

0x00C2 pbv at: rate: 

0x00C3 pbv at: 

0x00C4 pbv for: 

POB EXAMPLES 

 

HOMING TO A HARD STOP 

 

HOME_HARDSTOP 
// enter PBV mode 
pbv enter; 
// set torque limit to 2 Nm; save away previous torque limit 
_TLIM = LIM_TRQ; 
LIM_TRQ = 2; 
// start moving towards hard stop 
pbv at: 2 rate: 100; 
// wait until we hit the hard stop 
wait until OVER_TRQ; 
// wait a bit more to fully engage the hard stop 
wait 300 ms; 
// stop “here” and exit PBV mode 
pbv at: 0; 
pbv exit; 
// zero the following error 
zero following error; 
// call this position “0” 
set target position: 0; 
// restore torque limit 
LIM_TRQ = _TLIM; 
end; 

 

SIMPLE INDEXER 

 

INDEXER 
// Inputs #1 through #3 select the index length 
// Input #4 triggers the index 
[index] 
wait for rise of 4; 
// mask off the lower 3 bits of INS (that holds the state of the inputs) 
//    and use those bits to index into a vector array 
move trap for: VEC1[INS & 7]; 
wait for in position; 
goto index; 

FLYING SHEAR 

 

FLYING_SHEAR 
// Global variables: 
//   AD_DISP      accel/decel displacement (to gear up/down within) 
//   CUT_DISP     the cut displacement (pre-cut and post-cut) 
//   TIME_RET     the time to return to the home position 
//   CUT_LEN      the length of the product  
// 
// enter following mode 
follow enter; 
// clear the master dwell variable 
_MDWELL = 0; 
[cutloop] 
// clear the first slave and master counters 
clear (ASC1, AMC1); 
// gear up to line speed after some number of master counts have occurred 
gear at: 1 : 1 in: AD_DISP after: _MDWELL; 
// pre-cut displacement 
wait master CUT_DISP; 
// trigger the cut 
pulse output 1 for 100 ms; 
// post-cut displacement 
wait master CUT_DISP; 
// gear to 0 speed before we make our return to home 
gear at: 0 : 1 in: AD_DISP; 
// move back home 
//   home is magically ASC1 counts back, since ASC1 was cleared before the slave motion started 
move slave by: –ASC1 in: TIME_RET; 
// compute how long to dwell next time based upon the length of the material desired 
//   since CUT_LEN is the desired number of master counts we are supposed to cut to, and 
//   AMC1 holds how far we went this time, then the difference is how much we should wait 
//   before starting the next gear-up 
_MDWELL = CUT_LEN – AMC1; 
goto cutloop; 

ROTARY KNIFE 

 

ROTARY_KNIFE 
// Global variables: 
//   AD_DISP      accel/decel displacement (to gear up/down within) 
//   CUT_DISP     the cut displacement 
//   TIME_RET     the time to return to the home position 
//   CUT_LEN      the length of the product  
// 
// enter following mode 
follow enter; 
// clear the master dwell variable 
_MDWELL = 0; 
[cutloop] 
// clear the first slave modulo counter and the first master counter 
clear (MSC1, AMC1); 
// gear up to line speed after some number of master counts have occurred 
gear at: 1 : 1 in: AD_DISP after: _MDWELL; 
// cut displacement 
wait master CUT_DISP; 
// gear to 0 speed before we make our return to home 
gear at: 0 : 1 in: AD_DISP; 
// compute distance to the home position 
//   home is magically MODS-MSC1 counts forward, since MSC1 was cleared before the slave motion started 
_TOHOME = MODS – MSC1; 
move slave by: _TOHOME in: TIME_RET; 
// compute how long to dwell next time based upon the length of the material desired 
_MDWELL = CUT_LEN – AMC1; 
goto cutloop; 

LIMIT SWITCH HOMING 

 

POSLIM_HOMING 
// This POB homes an axis using a positive limit switch in conjunction with an index pulse 
// Operation is as follows: 
// 1. Head positive until the limit switch (one input 1) is encountered 
// 2. Stop, turn around and locate the first index pulse in the negative direction 
// 3. Stop and call that position 0.0 
// 
// enter PBV mode 
pbv enter; 
// start moving positive 
pbv at: 10; 
// wait until we see the limit switch 
wait for rise of 1; 
// and stop 
pbv stop within: 0.1; 
// exit PBV mode 
pbv exit; 
// move to index pulse (first in the negative direction is always held in TPOS_NZ) 
move trap to: TPOS_NZ; 
// wait until we’re there 
wait for in position; 
// call it 0.0 
set target position: 0; 
// done 
end; 

 


